ANSYS Mechanical APDL Rotordynamic Analysis Guide

ANSYS

ANSYS, Inc.
Southpointe
2600 ANSYS Drive
Canonsburg, PA 15317
ansysinfo@ansys.com
http://www.ansys.com
(T) 724-746-3304
(F) 724-514-9494

Release 18.1
April 2017

ANSYS, Inc. and ANSYS Europe, Ltd. are UL registered ISO 9001: 2008 companies.

Copyright and Trademark Information

Disclaimer Notice

U.S. Government Rights

Third-Party Software

Published in the U.S.A.

Table of Contents

List of Figures

Chapter 1: Introduction to Rotordynamic Analysis

Rotordynamics is the study of vibrational behavior in axially symmetric rotating structures. Devices such as engines, motors, disk drives and turbines all develop characteristic inertia effects that can be analyzed to improve the design and decrease the possibility of failure. At higher rotational speeds, such as in a gas turbine engine, the inertia effects of the rotating parts must be consistently represented in order to accurately predict the rotor behavior.

An important part of the inertia effects is the gyroscopic moment introduced by the precession motion of the vibrating rotor as it spins. As spin velocity increases, the gyroscopic moment acting on the rotor becomes critically significant. Not accounting for these effects at the design level can lead to bearing and/or support structure damage. Accounting for bearing stiffness and support structure flexibility, and then understanding the resulting damping behavior is an important factor in enhancing the stability of a vibrating rotor.

The modeling features for gyroscopic effects and bearing support flexibility are described in this guide. By integrating these characteristic rotordynamic features into the standard FEA modal, harmonic and transient analysis procedures found in ANSYS you can analyze and determine the design integrity of rotating equipment.

There are also specialized postprocessing features you can use to analyze specific behavior, and to process your simulation results to determine critical parameters. Orbit plots visualize the rotor's forward and backward whirl in a manner that allows you to easily determine the critical factors and the areas of concern. With the Campbell plots, you can determine critical speeds and system stability. These techniques, along with a number of other modeling and results analysis techniques are also covered in this guide.

Figure 1.1: Rotor Bearing System



Figure 1.2: Hard Disk Drive Mode Shape



The following additional topics offer more information to help you understand rotordynamics and how ANSYS supports rotordynamic analysis:

1.1. The General Dynamics Equations

The general dynamic equation is:

$$[M]\{\ddot{U}\}+[C]\{\dot{U}\}+[K]\{U\}=\{f\} \tag{1.1}$$

where [M], [C] and [K] are the mass, damping and stiffness matrices, and {f} is the external force vector.

In rotordynamics, this equation gets additional contributions from the gyroscopic effect [G], and the rotating damping effect [B] leading:

$$[M]\{\ddot{U}\}+([G]+[C])\{\dot{U}\}+([B]+[K])\{U\}=\{f\} \tag{1.2}$$

This equation holds when motion is described in a stationary reference frame, which is the scope of this guide.

The gyroscopic matrix, [G], depends on the rotational velocity (or velocities if parts of the structure have different spins) and is the major contributor to rotordynamic analysis. This matrix is unique to rotordynamic analyses, and is addressed specifically by certain commands and elements.

The rotating damping matrix, [B] also depends upon the rotational velocity. It modifies the apparent stiffness of the structure and can produce unstable motion.

For more information on those matrices, see Gyroscopic Matrix in the *Mechanical APDL Theory Reference*

1.2. The Benefits of the Finite Element Analysis Method for Modeling Rotating Structures

Rotating structures have conventionally been modeled by the lumped mass approach. This approach uses the center of mass to calculate the effects of rotation on attached or proximal components . A major limitation of this approach is the imprecise approximation of both the location and the distribution of the mass and inertias, along with the resulting inaccuracy in the calculation of internal forces and stresses in the components themselves.

The finite element (FE) method used in ANSYS offers an attractive approach to modeling a rotordynamic system. While it may require more computational resources compared to standard analyses, it has the following advantages:

- Accurate modeling of the mass and inertia
- A wide range of elements supporting gyroscopic effects
- The use of the CAD geometry when meshing in solid elements
- The ability of solid element meshes to account for the flexibility of the disk as well as the possible coupling between disk and shaft vibrations.
- The ability to include stationary parts within the full model or as substructures.

1.3. Overview of the Rotordynamic Analysis Process

A rotordynamic analysis involves most of the general steps found in any ANSYS analysis, as follows:

Step	Action	Comments
1.	Build the model.	A rotating structure generally consists of rotating parts, stationary parts, and bearings linking the rotating parts to the stationary parts and/or the ground. Understanding the relationships between these parts is often easier when the model is constructed to separate and define them. For more information about how to build the different parts, see Selecting and Components in the *Basic Analysis Guide*
2.	Define element types.	The elements that you select for the rotating parts of your model must support gyroscopic effects. The **CORIOLIS** command documentation lists the elements for which the gyroscopic matrix is available. All rotating parts must be axisymmetric. Model the stationary parts with any of the 3-D solid, shell, or beam elements available in the ANSYS element library. You can also add a stationary part as a substructure. For more information about how to generate and use a superelement, see Benefits of Substructuring in the *Substructuring Analysis Guide*. Model the bearings using either a spring/damper element COMBIN14, a general stiffness/damping matrix MATRIX27, a bearing element COMBI214, or a multipoint constraint element MPC184. MPC
3.	Define materials.	Defining the material properties for a rotordynamic analysis is no different than defining them in any other analysis. Use the **MP** or **TB** commands to define your linear and nonlinear material properties. See Defining Material Properties in the *Basic Analysis Guide*.
4.	Define the rotational velocity	Define the rotational velocity using either the **OMEGA** or **CMOMEGA** command. Use **OMEGA** if the whole model is rotating. Use **CMOMEGA** if there is a stationary parts and/or several rotating parts having different rotational velocities. **CMOMEGA** is based on the use of components, see Selecting and Components in the *Basic Analysis Guide*
5.	Account for gyroscopic effect	Use the **CORIOLIS** command to take into account the gyroscopic effect in all rotating parts as well as the rotating damping effect.
6.	Mesh the model.	Use the ANSYS meshing commands to mesh the parts. Certain areas may require more detailed meshing and/or specialized considerations. For more information, see the *Modeling and Meshing Guide*.
7.	Solve the model.	The solution phase of a rotordynamic analysis adheres to standard ANSYS conventions, keeping in mind that the gyroscopic matrices (as well as possibly the bearing matrices) may not be symmetric. Modal, harmonic and transient analyses can be performed. Performing several modal analyses allows you to review the stability and obtain critical speeds from the Campbell diagrams. A harmonic analysis allows you to calculate the response to synchronous (for example, unbalance) or asynchronous excitations.

Step	Action	Comments
		A transient analysis allows you to study the response of the structure under transient loads (for example, a 1G shock) or analyze the startup or stop effects on a rotating spool and the related components. Prestress can be an important factor in a typical rotordynamic analysis. You can include prestress in the modal, transient, or harmonic analysis, as described in the *Structural Analysis Guide* for each analysis type.
8.	Review the results.	Use POST1 (the general postprocessor) and POST26 (the time-history postprocessor) to review results. Specific commands are available in POST1 for Campbell diagram analysis (**PLCAMP**, **PRCAMP**), animation of the response (**ANHARM**) and orbits visualization and printout (**PLORB**, **PRORB**).

Chapter 2: Rotordynamic Analysis Tools

This section lists the primary commands and elements you will use in your rotordynamics analysis, along with reference materials.

The following topics are covered:

2.1. Commands Used in a Rotordynamic Analysis

The following commands are commonly used when performing a rotordynamic analysis:

Solver commands (/SOLU)	
CAMPBELL	Prepares the result file for a subsequent Campbell diagram of a prestressed structure.
CMOMEGA	Specifies the rotational velocity of an element component about a user-defined rotational axis.
CORIOLIS	Applies the gyroscopic effect to a rotating structure. Also applies the rotating damping effect.
OMEGA	Specifies the rotational velocity of the structure about global Cartesian axes.
SYNCHRO	Specifies whether the excitation frequency is synchronous or asynchronous with the rotational velocity of a structure in a harmonic analysis.

Postprocessing commands (/POST1)	
ANHARM	Produces an animation of time-harmonic results or complex mode shapes.
PLCAMP	Plots Campbell diagram data.
PLORB	Displays the orbital motion.
PRCAMP	Prints Campbell diagram data as well as critical speeds.
PRORB	Prints the orbital motion characteristics.

2.2. Elements Used in a Rotordynamic Analysis

Elements that are part of the rotating structure must account for the gyroscopic effect induced by the rotational angular velocity. The **CORIOLIS** command documentation lists the elements for which the gyroscopic matrix is available.

For information about current element technologies, see Legacy vs. Current Element Technologies in the *Element Reference*.

2.3. Terminology Used in a Rotordynamic Analysis

The following terms describe rotordynamic phenomena:

2.3.1. Gyroscopic Effect

For a structure spinning about an axis Δ, if a rotation about an axis perpendicular to Δ (a precession motion) is applied to the structure, a reaction moment appears. That reaction is the gyroscopic moment. Its axis is perpendicular to both the spin axis Δ and the precession axis.

The resulting gyroscopic matrix, [G], couples degrees of freedom that are on planes perpendicular to the spin axis. It is skew symmetric.

2.3.2. Whirl

When a rotating structure vibrates at its resonant frequency, points on the spin axis undergo an orbital motion, called whirling. Whirl motion can be a forward whirl (**FW**) if it is in the same direction as the rotational velocity or backward whirl (**BW**) if it is in the opposite direction.

2.3.3. Elliptical Orbit

In the most general case, the steady-state trajectory of a node located on the spin axis, also called orbit, is an ellipse. Its characteristics are described below.

In a local coordinate system xyz where x is the spin axis, the ellipse at node I is defined by semi-major axis A, semi-minor axis B, and phase ψ (PSI), as shown:

Figure 2.1: Elliptical Orbit



Angle φ (PHI) defines the initial position of the node (at t = 0). To compare the phases of two nodes of the structure, you can examine the sum $\psi + \varphi$.

Values YMAX and ZMAX are the maximum displacements along y and z axes, respectively.

2.3.4. Stability

Self-excited vibrations in a rotating structure cause an increase of the vibration amplitude over time such as shown below.

Figure 2.2: Instability



Such instabilities, if unchecked, can result in equipment damage.

The most common sources of instability are:

- Bearing characteristics (in particular when nonsymmetric cross-terms are present)
- Internal rotating damping (material damping)
- Contact between rotating and static parts

2.3.5. Critical Speed

The critical speed is the rotational speed that corresponds to the structure's resonance frequency (or frequencies). A critical speed appears when the natural frequency is equal to the excitation frequency. The excitation may come from unbalance that is synchronous with the rotational velocity or from any asynchronous excitation.

Critical speeds can be determined by performing a Campbell diagram analysis, where the intersection points between the frequency curves and the excitation line are calculated.

Critical speeds can also be determined directly by solving a new eigenproblem, as follows. First, the case of a single rotor is detailed, then the more general case of multiple rotating and/or stationary parts.

2.3.5.1. Direct Critical Speeds Calculation for a Single Rotor

For an undamped rotor, the dynamics equation (Equation 1.1 (p. 3)) is rewritten as:

$$[M]\{\ddot{U}\}+\omega[G_1]\{\dot{U}\}+[K]\{U\}=0 \quad (2.1)$$

Where $[G_1]$ is the gyroscopic matrix corresponding to a unit rotational velocity ω.

The solution is sought in the form:

$$\{U\}=\{\phi\}e^{j\bar{\lambda}t} \tag{2.2}$$

Where $\{\Phi\}$ is the mode shape and $\bar{\lambda}$ is the natural frequency.

The critical speeds are natural frequencies that are proportional to the rotational velocity.

The proportionality ratio α is defined as:

$$\bar{\lambda}=\alpha\omega \tag{2.3}$$

If the excitation is synchronous (for example, unbalanced excitation), the proportionality ratio is equal to 1.0.

Replacing Equation 2.2 (p. 11) and Equation 2.3 (p. 11) into Equation 2.1 (p. 10) leads to the new eigenproblem:

$$([K]-\bar{\lambda}^2[\bar{M}])\{\phi\}=0 \tag{2.4}$$

Where:

$$[\bar{M}]=[M]-j\frac{1}{\alpha}[G_1] \tag{2.5}$$

This problem can be solved using the unsymmetric eigensolver (**MODOPT**,UNSYM) along with the APDLMath command ***EIGEN**. See Eigenvalue and Eigenvector Extraction for more details about ANSYS eigensolvers.

2.3.5.2. Direct Critical Speeds Calculation for Multiple Rotating and/or Stationary Parts

For an undamped multi-rotor system with stationary parts, the dynamics equation (Equation 1.1 (p. 3)) is rewritten as:

$$[M]\{\ddot{U}\}+\sum_{i=1}^{N}\omega_i[G_1]_i\{\dot{U}\}+[K]\{U\}=0 \tag{2.6}$$

where:

N = total number of rotating and stationary parts

$[G_1]_i$ = unit gyroscopic matrix of the i^th part. If the i^th part is stationary, $[G_1]_i$ is zero.

ω_i = the rotational velocity of the i^th part. If the i^th part is stationary, ω_i is zero.

Assuming the first part is rotating, and taking it as the reference rotor ($\omega=\omega_1$), rotating parts' spins are expressed as:

$$\omega_i=a_i+b_i\omega \tag{2.7}$$

where:

a_i,b_i = relative spin rates of the i^th part. Note that for the reference rotor $a_1=0$ and $b_1=1$. For stationary parts, $a_1=b_1=0$.

Replacing Equation 2.7 (p. 11), Equation 2.3 (p. 11), and Equation 2.2 (p. 11) in Equation 2.6 (p. 11), a new eigenproblem is obtained:

$$([K]+j\bar{\lambda}[\bar{C}]-\bar{\lambda}^2[\bar{M}])\{\phi\}=0 \tag{2.8}$$

where:

$$[\overline{C}]=\sum_{i=1}^{N} a_i[G_1]_i$$

$$[\overline{M}]=\left([M]-j\sum_{i=1}^{N} b_i[G_1]_i\right)$$

This problem can be solved using the damped eigensolver (**MODOPT**,DAMP) along with the APDLMath command ***EIGEN**. See Eigenvalue and Eigenvector Extraction for more details about ANSYS eigensolvers.

2.3.6. Critical Speed Map

When designing a rotor, it is important to understand the effect of the bearing stiffness on the critical speeds. The critical speed map can be used to show the evolution of the critical speeds of the rotor with respect to the bearings stiffness.

To directly generate this map, the eigenproblem defined in Equation 2.4 (p. 11) is solved for different values of bearing stiffness. In this case, the model is considered undamped with identical bearings. See example in Example: Critical Speed Map Generation (p. 69).

2.4. Rotordynamics Reference Sources

In addition to the documentation for the commands (p. 7) and elements (p. 7) used in a rotordynamic analysis, other sources of information are available to help with your analysis.

2.4.1. Internal References

Although this guide is specific to rotordynamic applications, you can refer to the following ANSYS, Inc. documentation for more information about rotordynamics and related rotational phenomena:

Understanding Rotating Structure Dynamics in the *Advanced Analysis Guide*
Gyroscopic Matrix in the *Mechanical APDL Theory Reference*
Rotating Structures in the *Mechanical APDL Theory Reference*

The *Mechanical APDL Verification Manual* contains the following rotordynamics cases:

VM247 - Campbell Diagrams and Critical Speeds Using Symmetric Bearings
VM254 - Campbell Diagrams and Critical Speeds Using Symmetric Orthotropic Bearings
VM261 - Rotating Beam With Internal Viscous Damping

2.4.2. External References

A considerable body of literature exists covering the phenomena, modeling, and analysis of rotating structure vibrations. The following list of resources provides a good foundation for the subject:

D. Childs. Turbomachinery Dynamics. John Wiley 1993.

M. Lalanne and G. Ferraris. Rotordynamics Prediction in Engineering. John Wiley 2nd edition 1998.

G. Gienta. Dynamics of Rotating Systems. Springer 2005

H.D. Nelson and J.M. Mc Vaugh. The dynamics of rotor-bearing systems using finite elements. Journal of Engineering For Industry. May 1976. ASME.

M.Geradin and N. Kill. A new approach to finite element modelling of flexible rotors. Engineering Computations. March 1984

J. S. Rao. Rotor Dynamics. Wiley Eastern. India. 1985.

Chapter 3: Modeling a Rotordynamic Analysis

General Modeling and Meshing information can be found in the *Modeling and Meshing Guide*. This section contains the following topics to help you optimize model construction using the appropriate elements:

3.1. Building the Model

When building a model in an analysis involving rotordynamics, it is important to identify and separate rotating and non-rotating parts to:

- Apply the rotational velocity (or velocities) to the rotating parts
- Make sure the rotating parts are axisymmetric

Whether you construct your model in ANSYS, or you import it from an external CAD program, you will want to use the grouping and selecting capabilities in ANSYS to define areas of your model in ways that will optimize your analysis.

In the case of a rotordynamic analysis, this means identifying the spool, the bearings, the support structure and other areas as components or assemblies. See Selecting and Components in the *Basic Analysis Guide* for more information on how this capability can be applied to your analysis.

3.2. Selecting Parts and Bearings

To model a rotordynamic analysis, you must select appropriate elements for the parts and bearings.

Parts

A rotordynamic analysis model consists of rotating and stationary parts:

- The rotating parts are modeled using elements which support the gyroscopic effect. See Elements Used in a Rotordynamic Analysis (p. 7) for a list of elements.
- You can use any element type including superelements (MATRIX50) for non-rotating parts.

Bearings

To model bearings by inputting the stiffness and damping characteristics, select the most appropriate element type for your application from the following table.

	Description	Stiffness and Damping cross terms	Nonlinear stiffness and damping characteristics
COMBIN14	Uniaxial spring/damper	None	None

COMBI214	2-D spring/damper	Unsymmetric	Function of the rotational velocity
MATRIX27	General stiffness or damping matrix	Unsymmetric	None
MPC184	Multipoint constraint element	Symmetric for linear characteristics None for nonlinear characteristics	Function of the displacement

You can also model a plain cylindrical journal bearing or squeeze film damper with COMBI214 (KEYOPT(1) > 0) or FLUID218 (KEYOPT(1) = 1) which integrates the Reynolds equation for thin fluid film. Using the COMBI214 element, the stiffness and damping characteristics can be determined by performing a static analysis and specifying a perturbation increment in the element real constants. The characteristics are internally calculated based on a small perturbation about the equilibrium position.

Both COMBI214 and FLUID218 elements can also be used in a nonlinear large deflection transient analysis where the fluid film pressure forces are calculated at each time step.

The following topics provide more information about the element options listed in the table:

3.2.1. Using the COMBIN14 Element

The COMBIN14 element allows stiffness and/or damping characteristics in one direction. To define a bearing with characteristics KX and CX along X axis:

```
KX = 1.e+5          !Example stiffness value
CX = 100            !Example damping value

et,1,combin14
keyopt,1,2,1        ! X direction
r,1,KX,CX
```

KEYOPT(2) must be specified to define the active degree of freedom. This element operates in the nodal coordinate system.

3.2.2. Using the COMBI214 Element

The 2-D element supports user-defined stiffness/damping characteristics (KEYOPT(1) = 0) or the calculation of the bearing characteristics (or the bearing forces) for a plain cylindrical journal bearing based on finite length assumption (KEYOPT(1) > 0).

3.2.2.1. User-Defined Stiffness and Damping Characteristics (KEYOPT(1) = 0)

The COMBI214 element allows stiffness and/or damping characteristics in 2 perpendicular directions as well as cross-terms. To define a bearing in the YZ plane:

```
et,1,combi214
keyopt,1,2,1                          ! YZ plane
```

```
r,1,KYY,KZZ,KYZ,KZY,CYY,CZZ
rmore,CYZ,CZY
```

Note

KEYOPT(2) must be specified to define active degrees of freedom. This element operates in the nodal coordinate system.

In the case of a hydrodynamic bearing for example, the characteristics may vary with the rotational velocity. In this case, you need to specify OMEGS as the table parameter primary variable (***DIM** command). It is supported when activating the **CORIOLIS** command in a modal analysis (**ANTYPE**,MODAL), full harmonic analysis (**ANTYPE**,HARMIC), or full transient analysis (**ANTYPE**,TRANS).

An example of varying characteristics KYY and KZZ is given below:

```
et,1,combi214
keyopt,1,2,1                        ! YZ plane

! define table KYY
*DIM,KYY,table,3,1,1,omegs           ! table of dimension 3 depending upon omegs
KYY(1,0) = 0 , 1000 , 2000           ! 3 rotational velocities (rd/s)
KYY(1,1) = 1.e+6 , 2.7e+6 , 3.2e+6   ! stiffness characteristic at each rotational velocity

! define table KZZ
*DIM,KZZ,table,3,1,1,omegs           ! table of dimension 3 depending upon omegs
KZZ(1,0) = 0 , 1000 , 2000           ! 3 rotational velocities (rd/s)
KZZ(1,1) = 1.4e+6 , 4.e+6 , 4.2e+6   ! stiffness characteristic at each rotational velocity

r,1,%KYY%,%KZZ%
```

Note

If the characteristics of the COMBI214 element are varying with the rotational velocity and if the component rotational velocities are used (**CMOMEGA**), make sure the element is part of the appropriate rotating component.

In the case of a squeeze film damper, the characteristics may vary with the rotor eccentricity and/or the phase shift between the rotor displacements in the two nodal directions. In this case, you need to specify ECCENT and/or THETA as the table parameter primary variables (***DIM** command). A basic example of varying characteristics KXX is given below:

```
*dim, KXX,table,2,2,, eccent, theta
KXX(1,0) = 0, 1.e-2  ! 2 eccentricity values
KXX(0,1) = -180   ! 2 theta values (in degree)
KXX(0,2) =  180
KXX(1,1) = 1.e+5, 1.e+4  ! stiffness values for each
KXX(1,2) = 1.e+6, 1.e+5  !  value of eccentricity and theta (2x2)
```

The characteristics can be imported directly from an ASCII text file using the APDL macro `importbearing1.mac`. The format of this text file is described in Appendix A: Bearing Characteristics File Format (p. 127). An example of the macro usage is shown below:

```
! Import the bearing characteristics read in file bearingAP.txt
!  and create the table parameters K11_3, K12_3...
importbearing1, 'bearingAP', 3
! Define the bearing element real constants
r,1, %K11_3%, %K22_3%,%K12_3%,%K21_3%,%C11_3%,%C22_3%,
rmore, %C12_3%, %C21_3%
```

3.2.2.2. Calculation of the Bearing Characteristics (KEYOPT(1) > 0)

In a static analysis, the stiffness and damping characteristics of a plain cylindrical journal bearing or squeeze film damper can be calculated using a small perturbation near a user-defined equilibrium position.

Example inputs are shown in Example: Calculation of a Plain Cylindrical Journal Bearing Characteristics (p. 85) and Example: Calculation of a Squeeze Film Damper Characteristics (p. 87).

These characteristics can then be used as COMBI214 real constants (KEYOPT(1) = 0) in a subsequent modal or harmonic analysis.

3.2.2.3. Calculation of the Nonlinear Bearing Forces (KEYOPT(1) > 0)

In a nonlinear large-deflection transient analysis, the bearing forces are calculated based on the bearing definition (real constants and material viscosity) and the instantaneous displacements and velocities.

A simple example input is shown in Example: Transient Analysis of a Plain Cylindrical Journal Bearing (p. 89).

Note

A bearing element usually exhibits large stiffness values to be able to support the rotating structure. Also, the bearing clearance is very small and fluid film pressure is high, hence very small displacements induce a change in the bearing forces. When running such an analysis, make sure the time step is small enough to represent the nonlinearity of the bearing.

3.2.3. Using the FLUID218 Element

The FLUID218 element is based on the integration of the Reynolds equation for thin fluid film in a plain cylindrical journal bearing. Unlike COMBI214 and because it is a 3-D element, any type of groove or supply hole can be modeled specifying pressure boundary conditions. It supports:

- Pressure degree of freedom only (KEYOPT(1) = 0) to be used in a static analysis where the pressure distribution and pressure forces are determined. See Example: Calculation of the Pressure Profile of a Plain Cylindrical Journal Bearing with Supply Orifice (p. 94).

- Pressure and structural degrees of freedom (KEYOPT(1) = 1) to be used in a nonlinear large-deflection transient analysis where the time-dependent shaft position, pressure, and pressure forces are calculated. The note in Calculation of the Nonlinear Bearing Forces (KEYOPT(1) > 0) (p. 18) about the specificity of such a nonlinear analysis applies. See Example: Transient Analysis of a Plain Cylindrical Journal Bearing (3-D Approach) (p. 100).

3.2.4. Using the MATRIX27 Element

The MATRIX27 element allows the definition of 12 x 12 stiffness and damping matrices. Those matrices can be symmetric or not.

Example ofMATRIX27 use:

```
et,1,matrix27,,2,4,1                    ! unsymmetric [K] with printout
et,2,matrix27,,2,5,1                    ! unsymmetric [C] with printout

! define stiffness matrix
```

```
KXX = 8.e+7  $ KXY = -1.e7          ! $ sign allows several commands on
KYX = -6.e+7  $ KYY = 1.e+8          ! the same line

r,1,  KXX,KXY               $ rmore,-KXX,-KXY
rmore,KYX,KYY               $ rmore,-KYX,-KYY
*do, ir, 1, 8
    rmore                              ! define zero values
*enddo
rmore,-KXX,-KXY             $ rmore,KXX,KXY
rmore,-KYX,-KYY             $ rmore,KYX,KYY

! define damping matrix
CXX = 8.e+3                 $ CXY = -3.e+3
CYX = -3.e+3                $ CYY = 1.2e+4

r,2,  CXX,CXY               $ rmore,-CXX,-CXY
rmore,CYX,CYY               $ rmore,-CYX,-CYY
*do, ir, 1, 8
    rmore                              ! define zero values
*enddo
rmore,-CXX,-CXY             $ rmore,CXX,CXY
rmore,-CYX,-CYY             $ rmore,CYX,CYY
```

3.2.5. Using the MPC184 General Joint Element

The MPC184 is a joint element with elastic stiffness and damping behavior. The characteristics are defined as 6 X 6 matrices using **TB** commands.

Example of MPC184 use:

```
keyopt,2,4,1             ! no rotations

sectype,2,joint,gene
local,11,0,4,0,0,0,0,0   ! coordinate system forming the joint element
secjoin,,11

KYY = 1.e+8
CYY = 1.e+6
KZZ = 1.e+10
CZZ = 1.e+2

tb,join,2,,,stiff
tbdata,7,KYY
tbdata,12,KZZ

tb,join,2,,,damp
tbdata,7,CYY
tbdata,12,CZZ
```

3.3. Modeling Hints and Examples

The following modeling hints and examples can help you to create the model for your rotordynamic analysis:

3.3.1. Adding a Stationary Part

The stationary portion of your model could be a housing, a fixed support, or a flange. To add a stationary part, first create the part mesh. Since the rotational velocity is applied only to the rotating part of the structure, you need to create a component based on the elements of the rotating parts.

An example input to create a rotating component and apply the component rotational velocity using the **CMOMEGA** command follows:

```
! create the model

! create the rotating component
esel,,type,,1,2
cm,RotatingPart,elem
allsel

! apply rotational velocity to rotating component only
cmomega,RotatingPart,1000.
```

3.3.2. Transforming Non-Axisymmetric Parts into Equivalent Axisymmetric Mass

If your model comprises a non-axisymmetric part, you can transform it into an equivalent axisymmetric mass using the following procedure.

- First select the non-axisymmetric part volumes using **VSEL** command
- Enter the **VSUM** command to printout global mass properties of these volumes.
- Delete all the volumes.
- Define a new mass element (MASS21) on a node located at the center of gravity of the volumes. Real constants are the calculated mass and rotary inertia properties. These characteristics are approximated to obtain the axisymmetry. For example if the rotational velocity axis is along X, then the mass in Y and Z directions, along with the rotary inertia YY and ZZ are equal.
- Define a rigid region between the mass element node and the rest of the structure using the **CERIG** command.

You can obtain more precise mass, center of mass and moments of inertia by using inertia relief calculations. For more information, see Mass Moments of Inertia in the *Mechanical APDL Theory Reference*.

3.3.3. Defining Multiple Spools

To define several rotating parts, first create the part meshes. Since each part has a different rotational velocity, you need to define each part as a component based on the elements.

An example input to create two rotating components and apply the component rotational velocities using the **CMOMEGA** command follows:

```
! create the model

! create the first rotating component
esel,,type,,1,2
cm,RotatingPart1,elem

! create the second rotating component
esel,inve
cm,RotatingPart2,elem
allsel

! apply rotational velocities to rotating components
cmomega,RotatingPart1,1000.
cmomega,RotatingPart2,3000.
```

3.3.4. Using Component Mode Synthesis (CMS) for Rotating Parts

The component mode synthesis (CMS) procedure can be used to model rotating parts in a rotordynamics analysis.

The following are the primary modeling steps:

When the CMS procedure is used, outputs specific to rotating structure dynamics are limited to:

- The animation of the whirl (**ANHARM**) which is available before and after the CMS expansion pass
- The Campbell diagram (**PRCAMP** and **PLCAMP**) which is available only before the expansion pass

Orbits (**PRORB** and **PLORB**) are not available.

3.3.4.1. Generation Pass (Creating the Superelement)

In this step, the rotor is reduced to two CMS superelements. The following example input illustrates the generation pass:

```
! Generation Pass (Creating the Superelement)

/filname,part
/solu
antype,substr
seopt,part,3,1
cmsopt,fix,40


cmomega,comp1,1000     ! Rotational velocity of component ‘comp1’ (previously defined) is 1000rd/s about global X-D
coriolis,on,,,on       ! Coriolis on, in a stationary reference frame

cmsel,s,comp1
cmsel,s,interface2
m,all,all
nsle

solve
allsel,all,all
save
fini
```

You can specify the rotational velocity of the rotating superelement using the **OMEGA** and **CMOMEGA** commands. The Coriolis effect is included using the **CORIOLIS** command in a stationary reference frame (**CORIOLIS**,ON,,,ON). In the generation pass, only gyroscopic damping can be used. Other types of damping such as element damping and Rayleigh damping (**DMPRAT**, **ALPHAD**, **BETAD**, **MP**,ALPD, **MP**,BETD, **TB**,SDAMP,,,,ALPD, and **TB**,SDAMP,,,,BETD) are not supported. Damping is defined in the use pass step.

3.3.4.2. Use Pass

Define a superelement by reading in the superelement matrices using the **SE** command to build the full model and perform the desired analysis.

The following example input illustrates reading superelement matrices in the use pass, and a modal analysis procedure:

```
! Use Pass

/filname,use
/prep7

et,1,matrix50
type,1
se,part

save
fini

! Perform Modal Analysis
/solu

antype,modal
modopt,QRDAMP,20,,,ON
mxpand,20,,,yes

solve
fini
```

In the use pass, you can specify damping using **DMPRAT**, **ALPHAD**, **BETAD**, **MP**,ALPD and **MP**,BETD. If non-rotating parts like bearings and supporting structure are present, they are modeled during the use pass. The **CMOMEGA** command is used to specify which parts are rotating and which are not.

The following example input illustrates modeling of a rotating superelement and a non-superelement in the use pass:

```
! Generation Pass

/filname,part
/solu
antype,substr
seopt,part,3,1
cmsopt,fix,Nmode

cmomega,comp1,1             ! Rotational velocity of ‘comp1’ (previously defined) is 1rd/s about global X-Dir (defi
coriolis,on,,,on            ! Coriolis on, in a stationary reference frame

cmsel,s,comp1
cmsel,s,interface2
m,all,all
nsle

solve
allsel,all,all
save
fini

! Use Pass

/filname,use
/prep7

!Superelement
et,1,matrix50
type,1
se,part

! Non-superelements mp,ex,1,2.0e11
mp,dens,1,7800
mp,nuxy,1,0.3

et,2,188
type,2
secnum,2
sectype,2,beam,ctube
secdata,0.20,0.30,1,16
```

```
k,3,3,0,0
k,4,7,0,0

l,3,4
lesize,1,,,8

type,2
secnum,2
mat,1
lmesh,1
allsel,all,all

et,4,214                    ! Creation of symmetric Bearing
keyopt,4,2,1
keyopt,4,3,0
real,4
r,4,1.0e6,1.4e5,,,10,10
rmore,,,,

n,1000,5,0.22,0

type,4
real,4
e,node(5,0,0),1000
allsel,all

nsel,s,loc,x,0
nsel,a,loc,x,10
d,all,all,0
nsel,all

d,1000,all,0
allsel,all

cpintf,all
esel,s,type,,2
cm,comp2,elem               ! Creation of non-superelement component
allsel,all,all

save
fini

! Perform Modal Analysis
/solu
antype,modal
modopt,qrdamp,20,,,on
mxpand, 20,,,yes
cmomega,comp1,1000,0,0      ! Rotational velocity of ‘comp1’ is 1000 rd/s about global X-Dir

cmomega,comp2,1000,0,0      ! Rotational velocity of ‘comp2’ is 1000 rd/s about global X-Dir
coriolis,on,,,on

solve
fini
```

In general, you should specify a unit rotational velocity vector in the generation pass and then specify the true rotational velocity during the use pass. The Coriolis effect must be activated during both the generation and use passes in this case (**CORIOLIS**,ON,,,ON). Internally, the reduced gyroscopic matrix from the generation pass is scaled with the amplitude of the rotational velocity vector defined during the use pass. This procedure is used in particular for the generation of a Campbell diagram where modal analyses are performed for different rotational velocities.

The following example input illustrates the scaling of the gyroscopic damping matrix to print Campbell diagram data:

```
! Generation Pass (Creating the Superelement)

/filname,part
/solu
```

```
antype,substr
seopt,part,3,1
cmsopt,fix,40


cmomega,comp1,1.0            ! Component ‘comp1’ (defined previously) is rotating about global X-Dir
coriolis,on,,,on             ! Coriolis on, in a stationary reference frame

cmsel,s,comp1
cmsel,s,interface2
m,all,all
nsle

solve
allsel,all,all
save
fini

!Use Pass

/filname,use
/prep7

et,1,matrix50
type,1
se,part

esel,s,type,,1
cm,rot1,elem
allsel,all,all

save
fini

!Perform Modal Analysis
/solu

antype,modal
modopt,qrdamp,20,,,ON
mxpand,20,,,yes

coriolis,on,,,on             ! Coriolis on, in a stationary reference frame

! First rotational velocity
cmomega,rot1,1000            ! Component ‘rot1’  is rotating at 1000rd/s about global X-Dir (Scaling)
solve

! Second rotational velocity
cmomega,rot1,2000            ! Component ‘rot1’  is rotating at 2000rd/s about global X-Dir (Scaling)
solve

fini

/post1

prcamp,,,,,rot1              ! Print Campbell diagram information for component ‘rot1’
fini
```

If the true rotational velocity is specified in both the generation and use passes, the resulting gyroscopic effect will be based on the square of the rotational velocity, due to an internal scaling of the gyroscopic matrix. Make sure you use unit rotational velocity during the generation pass to ensure the correct scaling.

To compute and postprocess quantities using nodal velocities and nodal accelerations (damping force, inertial force, kinetic energy, etc.) resulting from the gyroscopic effect, the **OUTRES** command with *DSUBres* = ALL must be issued in the first load step of the use pass. These quantities are not computed if:

- At least one superelement load vector is applied in the use pass (**SFE**,,,SELV)
- The use pass analysis is a mode-superposition method (**TRNOPT**,MSUP and **HROPT**,MSUP)

For more information, see Component Mode Synthesis in the *Substructuring Analysis Guide*.

3.3.4.3. Expansion Pass

This step is the same as described in The CMS Use and Expansion Passes in the *Substructuring Analysis Guide*.

Chapter 4: Applying Loads and Constraints in a Rotordynamic Analysis

After you have built your model, you can apply the loads and constraints. The general procedures found in Loading in the *Basic Analysis Guide* apply.

In a quasi-static analysis, the gyroscopic forces are calculated. See Applying Quasi-Static Loads (p. 27) for details.

In a rotordynamic analysis, rotating forces must be applied. See Defining Rotating Forces (p. 27) for details about how to define those forces in a transient or harmonic analysis.

4.1. Applying Quasi-Static Loads

In a static analysis, the gyroscopic forces are calculated from the gyroscopic matrix and instantaneous nodal velocities. Just like in a dynamics analysis, the gyroscopic matrix is generated from the rotational velocity input (**OMEGA** or **CMOMEGA**) and the activation of the **CORIOLIS** command. For more information, see Specifying Rotational Velocity and Accounting for the Gyroscopic Effect (p. 30).

The nodal velocities are specified using **IC** and **ICROTATE**. Depending on the finite elements used in your model, you may need one or both commands. For example, if your model is a line element model (6 degrees of freedom per node), because all nodes lie on the rotational velocity axis, use **IC** input to directly enforce the nodal rotational velocities (OMGX, OMGY, or OMGZ). If your model is a solid element model (3 degrees of freedom per node), use **ICROTATE** input to specify the rotational velocity and rotational velocity axis so as to generate equivalent nodal translational velocities. If your model is a shell element model (6 degrees of freedom per node), use **ICROTATE** and **IC** commands to specify both translational and rotational nodal velocities.

The centrifugal force due to **ICROTATE** definition is also included in the analysis.

4.2. Defining Rotating Forces

4.2.1. Rotating Forces in Transient Analysis

In a transient analysis, the rotating forces are defined using table array parameters to specify the amplitude of the forces in each direction, at each time step. The analysis example provided in Example: Transient Response of a Startup (p. 58) shows how this is accomplished.

4.2.2. Rotating Forces in Harmonic Analysis

4.2.2.1. Nodal Force

Because complex notations are used in a harmonic analysis, a rotating load is defined with both a real component and an imaginary component (as described in Harmonic Analysis for Unbalance or General Rotating Asynchronous Forces in the *Advanced Analysis Guide*). For example, to apply a rotating force F0 in the (YZ) plane, rotating in the counterclockwise direction (Y to Z), the force components are:

```
F0 = 1.e+6  ! sample force component value
INODE = node(0.1,0,0)  ! sample node number
F,INODE,fy, F0  ! real fy component at node INODE
F,INODE,fz,, -F0  ! imaginary fz component at node INODE
```

For more information, see Applying Loads and Obtaining the Solution in the *Structural Analysis Guide*.

If the rotating harmonic load is synchronous or asynchronous with the rotational velocity, use the **SYNCHRO** command. In this case, the amplitude of the force generated by unbalance represents the mass times the radius of the eccentric mass. The spin squared factor is introduced automatically. See Example: Unbalance Harmonic Response of a Two-Spool Rotor (p. 50) for more information about harmonic analysis with rotating forces.

4.2.2.2. Distributed Forces Coming From Solid or Shell Model Unbalance

If a solid or shell model is not exactly axisymmetric, it induces unbalance forces that may need to be taken into account in your harmonic analysis. To determine and apply such forces, do the following:

- Create nodes on the rotational velocity axis (center line) of the rotor model.
- Couple the model to the center nodes that define a rigid region (**CERIG** command).
- If the rotational velocity is along X, constrain UY, UZ, and ROTX degrees of freedom at the center nodes. Do not constrain the bearings locations.
- Perform a static analysis with unit rotational velocity (*OMEGX*=1.0 on the **OMEGA** command).
- In the post-processor (/POST1), retrieve and store the reaction forces at the center nodes (*Entity*=Node and *Item1*=RF on the ***GET** command).
- You can perform your harmonic analysis after you have:
  - Removed the constraints at the center nodes.
  - Added your constraints at the bearings locations.
  - Applied the stored reaction forces at the center nodes. Ensure that you take the opposite sign and define complex rotating forces.

Chapter 5: Solving a Rotordynamic Analysis

After modeling the structure and specifying the loads and constraints, you can run your rotordynamic analysis. Although certain differences will be covered in the subsequent sections, whether your analysis is modal, transient or harmonic the general procedures are very similar to those described in the solution portion of Applying Loads and Obtaining the Solution in the *Structural Analysis Guide*.

This following topics related to solving a rotordynamic analysis are available:

5.1. Adding Damping

Damping is present in most systems and should be specified for your dynamic analysis. Bearings are one of the most common sources of rotordynamic damping. More information on how to specify your bearing damping characteristics is found in Selecting Parts and Bearings (p. 15), also in this guide.

In addition, the following forms of damping are available:

- Alpha and Beta Damping (Rayleigh Damping) **ALPHAD BETAD**
- Material-Dependent Damping **MP**,BETD, **MP**,ALPD, **TB**,SDAMP,,,,BETD, **TB**,SDAMP,,,,ALPD
- Constant Material Damping Coefficient **MP**,DMPR
- Constant Damping Ratio **DMPRAT**
- Modal Damping **MDAMP**
- Element Damping
- Material Structural Damping Coefficient

See Damping in the *Structural Analysis Guide*. The accompanying tables provide more information on the types of damping that are available for your analysis.

The effect of rotating damping is specific to rotating structures. It is supported by all the elements which generate a gyroscopic matrix (see **CORIOLIS** for a list of elements) when proportional viscous or structural damping is defined. The viscous damping is defined via beta damping (**BETAD**) or the material-dependent damping (**MP**,BETD). The structural damping is defined via **DMPSTR** or **MP**,DMPR.

It is also supported by the following elements:

- Bearing element COMBI214 (real constants K11 = K22 and K21 = K12 = 0)

- Spring-damper element COMBIN14
- General joint MPC184 with 6 degrees of freedom (KEYOPT(1) = 16 and KEYOPT(4) = 0) with stiffness characteristics (**TB**,JOIN,,,,STIF)
- Stiffness matrix MATRIX27 (KEYOPT(2) = 0 and KEYOPT(3) = 4)

For COMBI214, the rotating damping effect can also come directly from the damping characteristics (real constants C11 = C22 and C21 = C12 = 0).

The rotating damping effect can be activated using the *RotDamp* argument of the **CORIOLIS** command. An example can be found in VM261 - Rotating Beam with Internal Viscous Damping.

5.2. Specifying Rotational Velocity and Accounting for the Gyroscopic Effect

The rotational velocity of the structure is specified via the **OMEGA** or **CMOMEGA** commands. For the **OMEGA** command, define the rotational velocity vector along one of the global coordinate system axes. The gyroscopic effect is set via the **CORIOLIS** command.

```
omega,1000.
coriolis, on,,, on   ! last field specifies stationary reference frame
```

Note

In rotordynamics, the effect of the rotating inertias is calculated in the stationary reference frame (the scope of this guide). The *RefFrame* argument of the **CORIOLIS** command must be set accordingly.

Unlike **OMEGA**, **CMOMEGA** lets you define a rotational velocity vector independent of the global X, Y or Z axes. For example, you may define the direction of the rotational velocity vector using two points and the rotational velocity as a scalar, as follows:

```
! Define rotational velocity for COMPONENT1:
! spin is 1000 rd/s
! direction is parallel to Z axis and passing through point (0.1,0,0)
cmomega, COMPONENT1, 1000.,,, 0.1, 0, 0, 0.1, 0,1
```

5.3. Solving for a Subsequent Campbell Analysis of a Prestressed Structure Using the Linear Perturbation Procedure

For more information on the Linear Perturbation procedure for rotating structures, see Considerations for Rotating Structures.

To generate the Campbell diagram of a prestressed structure, multiple static and perturbed modal solutions are calculated alternately. The Campbell action (*Action* on the **CAMPBELL** command) must be set to the results filename extension of the perturbed solution (RSTP) in the first static analysis.

The results of the modal analyses are appended to the `Jobname.RSTP` file to accommodate a subsequent Campbell diagram analysis – see Campbell Diagram (p. 39).

For an example analysis that includes a Campbell diagram of a thin disk, see Example: Campbell Diagram Analysis of a Prestressed Structure Using the Linear Perturbation Procedure (p. 48).

5.4. Solving a Harmonic Analysis with Synchronous or Asynchronous Rotating Forces

To perform a harmonic analysis of an unbalanced excitation, the effect of the unbalanced mass is represented by forces in the two directions perpendicular to the spinning axis. (See Defining Rotating Forces (p. 27).) The forces are applied on a node located on the axis of rotation. The **SYNCHRO** command is used to specify that the frequency of excitation is synchronous with the rotational velocity.

Note

The **SYNCHRO** command's *RATIO* argument is not valid in the case of an unbalanced force.

This linear approach can be used for beam models as well as for solid models.

For solid models, your analysis may require a more precise determination of displacements and stresses in the wheel/disk containing the unbalanced mass. In this case, you can model the unbalance using a MASS21 element and performing a nonlinear transient analysis.

5.4.1. Specifying Rotational Velocity with OMEGA

You can specify the rotational velocity using the **OMEGA** command. When the SYNCHRO command is activated, the OMEGA command defines the rotational velocity direction vector. The spin is specified automatically with the **HARFRQ** command. See the following example:

```
harfrq,100            ! 100 Hz
synchro
omega,1.,1.,1.        ! direction vector of the rotational velocity
```

The above commands denote:

- an excitation frequency of 100 Hz,
- a spin of (100) (2π) rd/sec
- a rotational velocity vector of

$$\omega_x=\omega_y=\omega_z=\frac{100*2*\pi}{\sqrt{3}}$$

5.4.2. Specifying Rotational Velocity with CMOMEGA

You can specify the rotational velocity using the **CMOMEGA** command. The rotational velocity definition is different when tabular input is used. See below for the details of tabular and non-tabular input.

5.4.2.1. Non-Tabular Input

When the **SYNCHRO** command is activated, the **CMOMEGA** command only defines the rotational velocity direction vector for the component. If there are several components, the ratios between their different spins are also calculated from the **CMOMEGA** input. The spin of the driving component (specified by Cname in the **SYNCHRO** command) is derived from the **HARFRQ** command, as noted in the following example:

```
harfrq,100.                ! excitation 100 Hz
synchro,,SPOOL1            ! driving component is SPOOL1
cmomega,SPOOL1,1.,1.,1     ! direction vector  of the rotational velocity for SPOOL1
cmomega,SPOOL2,2.,2.,2.    ! direction vector  of the rotational velocity for SPOOL2 (also spin ratio between the 2 c
```

The above commands denote:

- an excitation frequency of 100Hz
- the spin of SPOOL1 is (100) (2π) rd/sec, with a rotational velocity vector of:

$$\omega_x=\omega_y=\omega_z=\frac{100*2*\pi}{\sqrt{3}}$$

- the spin of SPOOL2 is twice the spin of SPOOL1 with the same rotational velocity vector

5.4.2.2. Tabular Input

Using tabular input may sometimes be necessary, for example when rotors' rotational velocities are not proportional. In this case, all rotating components must be defined with the **CMOMEGA** command and a tabular rotational velocity *OMEGAX*(%tab%) with primary variable FREQ.

For each harmonic analysis solution, the rotational velocity of the driving component is directly obtained from the excitation frequency and the two points provided. It is scaled with the asynchronous ratio if provided (*RATIO* on **SYNCHRO**).

For all other rotating components, the spin is calculated as the spin of the driving component multiplied by the ratio of the tabular inputs.

5.5. Selecting an Appropriate Solver

The solver you select depends on the analysis type, as follows:

5.5.1. Solver for a Modal Analysis

Both the DAMP and QRDAMP eigensolvers are applicable to a rotordynamic analysis. Before selecting an eigensolver, consider the following:

- If you intend to perform a subsequent modal superposition, harmonic or transient analysis, use the QRDAMP eigensolver. The DAMP eigensolver is not supported for mode-superposition methods.
- The DAMP eigensolver solves the full system of equations, whereas the QRDAMP eigensolver solves a reduced system of equations. Although the QRDAMP eigensolver is computationally more efficient than the DAMP eigensolver, it is restricted to cases where damping (viscous, material, etc.) is not critical.

Note

When using the QRDAMP eigensolver in a multiple load step modal analysis (a Campbell analysis for example), you can activate the *ReuseKey* on the **QRDOPT** command to reuse the eigenvectors from the symmetric eigensolution of the first step. This will result in better performances. Note, however, that if variable bearings are present (COMBI214 with tabular characteristics), this option may lead to incorrect results given the change in stiffness in each load step.

When rotating damping is included in the analysis (*RotDamp*=ON in the **CORIOLIS** command) and solid elements are used for the rotating parts of the structure, DAMP eigensolver is recommended.

After a complex modal analysis using the QRDAMP method, complex frequencies are listed in the following way:

```
 ***** DAMPED FREQUENCIES FROM REDUCED DAMPED EIGENSOLVER *****

   MODE         COMPLEX FREQUENCY (HERTZ)           MODAL DAMPING RATIO
     1     -0.78052954E-01       49.844724    j      0.15659202E-02
                (a)                 (b)                   (c)
```

where

(a) is the real part of the complex frequency. It shows the damping of this particular frequency as well as its stability. A negative real part reflects a stable mode while a positive one reflects an unstable mode. More information on instability can be found earlier in this guide under Stability (p. 9).
(b) is the complex part of the complex frequency. It represents the damped frequency.
(c) is the modal damping ratio. It is the ratio between the real part and the complex frequency modulus (also called norm of the complex frequency).

Although the gyroscopic effect creates a "damping" matrix, it does not dissipate energy; therefore, if there is no damping in a rotating structure, all the real parts of its complex frequencies are zero.

The complex part is zero if the complex frequency corresponds to a rigid body mode, or if it corresponds to an overdamped frequency where the damping is so high that it suppresses the vibration.

For more information, see Complex Eigensolutions in the *Mechanical APDL Theory Reference*

5.5.2. Solver for a Harmonic Analysis

The full method and the mode-superposition (based on QRDAMP modal analysis) method are supported for rotordynamic analyses.

If the **SYNCHRO** command is used (as in an unbalanced response calculation), the mode-superposition method is not supported. In this case, the gyroscopic matrix must be recalculated at each frequency step. Only the FULL method is applicable.

5.5.3. Solver for a Transient Analysis

In a small-deflection transient analysis, the Coriolis effect is activated using the **CORIOLIS** command. Full method and mode-superposition based on QRDAMP modal analysis method are supported for rotordynamics.

For the full method, use the Newton-Raphson with unsymmetric matrices option (**NROPT**, UNSYM).

If the rotational velocity is varying (as in the startup of a turbomachine), mode-superposition method is not supported. In this case, the gyroscopic matrix needs to be recalculated at each time step, and only the FULL method can be applied.

In a large-deflection transient analysis, (**NLGEOM**,ON), the Coriolis effect and all other nonlinear inertial effects are automatically included in the analysis and the **CORIOLIS** command should not be used. This is the case for rotor-bearing simulation using COMBI214 or FLUID218, for example.

5.6. Using Linear Perturbation Modal Analysis

If the effects of prestress need to be taken into account in your modal analysis, you may want to use linear perturbation analysis. For details about this procedure, see Considerations for Rotating Structures in the *Mechanical APDL Structural Analysis Guide*.

Chapter 6: Postprocessing a Rotordynamic Analysis

After you solve your analysis, you will want to analyze the results. This often involves processing data from the results file and organizing it so that the relevant parameters and their relationships are available. This section contains information on the tools you will use, along with examples of how to use them.

General information on postprocessing can be found in The General Postprocessor (POST1) and The Time-History Postprocessor (POST26) in the *Basic Analysis Guide*

The following specific topics are available here:

6.1. Postprocessing Complex Results

The results obtained from a modal or harmonic analysis are complex. They require specific postprocessing procedures detailed in POST1 and POST26 – Complex Results Postprocessing in the *Mechanical APDL Theory Reference*. The main procedures are given below.

6.1.1. In POST1

The general postprocessor POST1 allows you to review the solution at a specific excitation frequency after a harmonic analysis, or for a specific damped frequency after a complex modal analysis.

The SET command provides options to define the data set to be read from the results file. Specifically, the KIMG argument is used for complex results as follows:

- the real part (KIMG = REAL)
- the imaginary part (KIMG = IMAG)
- the amplitude (KIMG = AMPL)
- the phase (KIMG = PHAS)

It is also possible to store your solution at a given angle into the database using the **HRCPLX** command.

Once the desired data is stored in the database, you may use any postprocessing command to create graphics displays or tabular listings. See Reviewing Results in POST1 in the *Basic Analysis Guide* for more information.

6.1.2. In POST26

After a harmonic analysis, the time-history postprocessor (POST26) allows you to review your results at a specific location as a function of the frequency.

The general procedure for complex results processing follows that found in The Time-History Postprocessor (POST26) in the *Basic Analysis Guide*.

- Define your variables using the **NSOL**, **ESOL**, and **RFORCE** commands
- Process your variables to develop calculated data using the **ABS**, **IMAGIN**, **REALVAR** and **ADD** commands.
- Review the variables using the **PRVAR**, **PLVAR** and **EXTREM** commands.

When plotting complex data, **PLVAR** plots the amplitude by default. You can switch to plotting the phase angle or the real part or the imaginary part via the **PLCPLX** command.

When listing complex data, **PRVAR** printout the real and imaginary parts by default. You can switch to listing the amplitude and phase via the **PRCPLX** command.

6.2. Visualizing the Orbits After a Modal or Harmonic Analysis

To visualize the orbits after a modal or harmonic analysis has been performed, use the **PLORB** command in POST1.

Because the elliptical orbit is valid only for nodes on the rotational velocity axis, **PLORB** command is valid for current-technology beam, pipe or elbow elements. If you have a solid element model, you can add line elements (with negligible stiffness and mass) on the rotational velocity axis to visualize the orbits.

Sample command input to output your orbit plot at a given frequency:

```
/POST1
set,1,6          ! read load step 1, substep 6
plorb
```

The spool line is in dark blue, while the orbits are in light blue.

6.3. Printing the Orbit Characteristics After a Modal or Harmonic Analysis

To print out the characteristics of the orbits after a modal or harmonic analysis has been performed, use the **PRORB** command in /POST1. See Elliptical Orbit (p. 8) in this guide for a definition of the characteristics.

Because the elliptical orbit is valid only for nodes on the rotational velocity axis, the **PRORB** command is valid for current-technology beam, pipe or elbow elements. If you have a solid element model, you can add massless line elements on the rotational velocity axis so that the orbit characteristics are calculated and printed out.

The following command string prints out the orbit characteristics at a given frequency:

```
/POST1
set,1,6  ! read load step 1, substep 6
prorb
```

The angles are expressed in degrees for the range of -180° to +180°. The position vector of the local Y axis in the global coordinate system is printed out along with the elliptical orbit characteristics.

To retrieve and store your orbit characteristics as parameters, use the ***GET** command with `Item1` = ORBT after issuing the **PRORB** command.

6.4. Animating the Orbits After a Modal or Harmonic Analysis

To animate the orbits and visualize the whirling, use **ANHARM** command in /POST1. A sample input follows:

```
/POST1
set,1,6             ! read load step 1, substep 6
plnsol,u,sum        ! specify the results to be animated
anharm
```



6.5. Visualizing Your Orbits After a Transient Analysis

Plot the transient orbits using the **PLVAR** command, as shown in the following example:

```
/post26
INODE = 12                   ! node of interest
nsol,2,INODE,u,y                ! define variable 2
nsol,3,INODE,u,z                ! define variable 3

/axlab,X,displacement UY     ! specify Xaxis label
/axlab,Y,displacement UZ     ! specify Yaxis label

xvar,2                       ! variable 2 is on Xaxis
plvar,3                      ! plot variable 3 on Yaxis
```

6.6. Postprocessing Bearing and Reaction Forces

You can postprocess element forces only if those forces are written to the database. Database writing is controlled using the **OUTRES** command at the solver level. You may also printout the loads at the solver level using the **OUTPR** command.

To print out the reaction forces and element forces in the general postprocessor (/POST1):

```
/post1
set,last                ! last substep of last loadstep
```

```
! printout reaction forces
force,static          ! elastic forces (stiffness)
prrfor
force,damp            ! damping forces
prrfor

! printout element forces
force,static          ! elastic forces (stiffness)
presol,F
force,damp            ! damping forces
presol,F
```

If you use the COMBI214 or FLUID218 element to model the bearings, you can retrieve reaction forces from the element. Details on using this feature after your transient analysis follow.

6.6.1. COMBI214 Bearing Forces

Transient bearing reaction forces are part of element COMBI214 outputs. Elastic forces (also called spring forces) as well as damping forces are available along the principal axes of the element. All calculated forces include the cross-term effects.

You can use the POST26 time-history postprocessor to print out the stiffness and damping bearing forces, as shown in the following example:

```
/post26
! parameters for element and node number
BEARING_ELEM = 154
BEARING_NODE1 = 1005

! define elastic forces as variables 2 and 3
esol,2,BEARING_ELEM,BEARING_NODE1,smisc,1,FE1
esol,3,BEARING_ELEM,BEARING_NODE1,smisc,2,FE2

! damping forces as variables 4 and 5
esol,4, BEARING_ELEM,BEARING_NODE1,nmisc,5,FD1
esol,5, BEARING_ELEM,BEARING_NODE1,nmisc,6,FD2

! printout all forces as function of time
prvar,2,3,4,5

! plot all forces as function of time
plvar,2,3,4,5
```

6.6.2. FLUID218 Bearing Forces

Bearing forces are part of the FLUID218 output. Because it is a 3-D element, the element forces must be summed to obtain the total bearing forces. See Example: Calculation of the Pressure Profile of a Plain Cylindrical Journal Bearing with Supply Orifice (p. 94).

6.7. Campbell Diagram

After you have run a modal analysis with several rotational velocity load steps, you can perform a Campbell diagram analysis. The analysis allows you to:

- Visualize the evolution of the frequencies with the rotational velocity
- Check the stability (p. 9) and whirl (p. 8) of each mode
- Determine the critical speeds (p. 10).
- Determine the stability threshold

The plot showing the variation of frequency with respect to rotational velocity may not be readily apparent. For more information, see Generating a Successful Campbell Diagram (p. 42) below.

6.7.1. Visualize the Evolution of the Frequencies With the Rotational Velocity

In the general postprocessor (POST1), issue the **PLCAMP** command to display a Campbell diagram as shown below.



If there are rotating components, you will specify the name of the reference component via the `Cname` argument in the **PLCAMP** command.

A maximum of 10 frequency curves are plotted within the frequency range specified.

Use the following commands to modify the appearance of the graphics:

Scale
To change the scale of the graphic, you can use the **/XRANGE** and **/YRANGE** commands.

High Frequencies
Use the FREQB argument in the **PLCAMP** command to select the lowest frequency of interest.

Rotational Velocity Units
Use the UNIT argument in the PLCAMP command to change the X axis units. This value is expressed as either rd/sec (default), or rpm.

Use the SLOPE argument in **PLCAMP** command to display the line representing an excitation. For example, an excitation coming from unbalance corresponds to SLOPE = 1.0 because it is synchronous with the rotational velocity.

6.7.2. Check the Stability and Whirl of Each Mode

Forward (FW), and backward (BW) whirls, and unstable frequencies are identified in the Campbell diagram analysis. These characteristics appear in the Campbell diagram graphic legend generated by the **PLCAMP** command. Forward and backward whirls are printed out in the table generated by the **PRCAMP** command, as shown below.

```
PRINT CAMPBELL DIAGRAM
   Sorting : ON
   Slope of line :   1.000
   X axis unit : rpm



    ***** FREQUENCIES (Hz) FROM CAMPBELL (sorting on) *****


 Spin(rpm)          0.000      34999.970      69999.941     104999.911


 1.00xSpin          0.000        583.333       1166.666       1749.999
    1    BW       271.197        241.938        214.504        189.830
    2    FW       271.197        300.946        329.806        356.718
    3    BW       808.789        787.112        762.362        734.483
    4    FW       808.789        827.877        844.922        860.365
    5    BW      1272.346       1161.220       1068.659        996.570
    6    FW      1272.346       1394.676       1516.228       1624.442
    7    BW      2031.219       1994.892       1964.575       1938.020
    8    FW      2031.219       2076.671       2135.135       2209.887
    9    BW      2849.365       2647.757       2476.570       2331.084
```

If an unstable frequency is detected, it is identified in the table by a letter u between the mode number and the whirl characteristics (BW/FW). In this example, all frequencies are stable.

By default, the **PRCAMP** command prints a maximum of 10 frequencies (to be consistent with the plot obtained via the **PLCAMP** command). If you want to see *all* frequencies, set `KeyALLFreq` = 1.

You can determine how a particular frequency becomes unstable by issuing the **PLCAMP** or **PRCAMP** and then specifying a stability value (STABVAL) of 1. You can also view the logarithmic decrements by specifying a STABVAL = 2 or 3.

To retrieve and store frequencies and whirls as parameters: Use the ***GET** command with `Entity` = CAMP and `Item1` = FREQ or WHRL. A maximum of 200 values are retrieved within the frequency range specified.

6.7.3. Determine the Critical Speeds

The **PRCAMP** command prints out the critical speeds for a rotating synchronous (unbalanced) or asynchronous force when SLOPE is input:

```
***** CRITICAL SPEEDS (rpm) FROM CAMPBELL (sorting on) *****

              Slope of line :      1.000

                    1          15494.634
                    2          17146.270
                    3          46729.086
                    4          50114.286
                    5          64924.847
                    6          95750.714
                    7               none
                    8               none
                    9               none
```

The critical speeds correspond to the intersection points between frequency curves and the added line F = sω (where s represents SLOPE > 0 as specified via **PRCAMP**).

Because the critical speeds are determined graphically, their accuracy depends upon the quality of the Campbell diagram. For example, if the frequencies show significant variations over the rotational velocity range, you must ensure that enough modal analyses have been performed to accurately represent those variations. For more information about how to generate a successful Campbell diagram, seeGenerating a Successful Campbell Diagram (p. 42) below.

To retrieve and store critical speeds as parameters: Use the ***GET** command with `Entity` = CAMP and `Item1` = VCRI. A maximum of 200 values are retrieved within the frequency range specified.

6.7.4. Determine the Stability Threshold

The **PRCAMP** command prints out the stability threshold of each mode when a zero line (*SLOPE*) is input and stability values (or logarithmic decrements) are post-processed (*STABVAL* = 1, 2, or 3).

Stability thresholds correspond to change of signs. Because they are determined graphically, their accuracy depends upon the quality of the Campbell diagram. For example, if the stability values (or logarithmic decrements) show significant variations over the rotational velocity range, you must ensure that modal analyses have been performed with enough load steps to accurately represent those variations.

To retrieve and store stability thresholds as parameters, use the ***GET** command with `Entity` = CAMP and `Item1` = VSTA.

6.7.5. Generating a Successful Campbell Diagram

To help you obtain a good Campbell diagram plot or printout, the sorting option is active by default (**PLCAMP**,ON or **PRCAMP**,ON). ANSYS compares complex mode shapes obtained at 2 consecutive load steps using the Modal Assurance Criterion (MAC). The equations used are described in POST1 - Modal Assurance Criterion (MAC) in the *Mechanical APDL Theory Reference*. Similar modes shapes are then paired. If one pair of matched modes has a MAC value smaller than 0.7, the following warning message is output:

```
*** WARNING ***
Sorting process may not be successful due to the shape of some modes.
If results are not satisfactory, try to change the load steps and/or
the number of modes.
```

If such a case, or if the plot is otherwise unsatisfactory, try the following:

- Start the Campbell analysis with a nonzero rotational velocity.

  Modes at zero rotational velocity are real modes and may be difficult to pair with complex modes obtained at nonzero rotational velocity.

- Increase the number of load steps.

  It helps if the mode shapes change significantly as the spin velocity increases.

- Change the frequency window.

  To do so, use the shift option (**PLCAMP**,,,*FREQB* or **PRCAMP**,,,*FREQB*). It helps if some modes fall outside the default frequency window.

Chapter 7: Rotordynamic Analysis Examples

The following example analysis samples are available:

7.1. Example: Campbell Diagram Analysis

To generate the Campbell diagram of a simply supported rotating beam, see Sample Campbell Diagram Analysis in the *Advanced Analysis Guide*

For the Campbell diagram and critical speed analysis of a rotor on bearings, see VM247 "Campbell Diagrams and Critical Speeds Using Symmetric Bearings" and VM254 "Campbell Diagrams and Critical Speeds Using Symmetric Orthotropic Bearings" in the *Mechanical APDL Verification Manual*.

For the Campbell diagram and stability analysis of a rotating beam on bearings with viscous internal damping, see VM261 "Rotating Beam with Internal Viscous Damping" in the *Mechanical APDL Verification Manual*.

The following section presents a Campbell diagram analysis of the clamped-free disk shown in Figure 7.1: Clamped Disk (p. 46).

The model is a thin disk with the inner radius clamped and the outer radius free. The rotational velocity is 120 Hz along the Z axis.

Figure 7.1: Clamped Disk



7.1.1. Problem Specifications

The geometric properties for this analysis are as follows:

Thickness: 0.8 mm
Inner radius: 16.5 mm
Outer radius: 47.5 mm

The material properties for this analysis are as follows:

Young's modulus (E) = 7.2e+10 N/m^2
Poisson's ratio (υ) = 0.3
Density = 2800 kg/m^3

7.1.2. Input for the Analysis

```
! ** parameters
pi = acos(-1)
xa = 47.5e-3
xb = 16.5e-3
zh = 0.8e-3
spin = 120*2*pi

/prep7
et,1,181
r,1,zh

! ** material = aluminium
mp,ex,,7.2e+10
mp,nuxy,,.3
mp,dens,,2800.

! ** mesh
esize,0.0025
cyl4,,,xb,0,xa,360
amesh,all

! ** constraints = clamp inner radius
lsel,,,,5,8
```

```
dl,all,1,all
allsel
fini

! *** modal analysis in rotation
/solu
antype,modal
modopt,qrdamp,30,,,on!! Cpxmod=on
qrdopt,on !! ReuseKey=on
mxpand,30
coriolis,on,,,on

omega,,,0.1  !! non zero to easy the Campbell diagram sorting
solve

omega,,,spin/2
solve

omega,,,spin
solve

finish

! *** campbell diagram
/post1
/yrange,500,1500
/show,JPEG
plcamp
/show,CLOSE
prcamp
finish
```

7.1.3. Output for the Analysis

Figure 7.2: Campbell Diagram for the Clamped Disk



Figure 7.3: Frequency Outputs for the Clamped Disk

```
      ***** FREQUENCIES (Hz) FROM CAMPBELL (sorting on) *****


  Spin(rd/s)           0.100          376.991          753.982

      1              659.204          602.520          550.666
      2              660.123          722.411          790.430
      3              660.725          660.539          660.565
      4              776.025          666.221          573.261
      5              777.819          906.038         1052.928
      6             1202.574         1036.442          895.790
      7             1203.444         1396.346         1615.597
```

7.2. Example: Campbell Diagram Analysis of a Prestressed Structure Using the Linear Perturbation Procedure

This problem is the same as the one described above, except that the effect of the prestress due to the centrifugal force is taken into account.

7.2.1. Input for the Analysis

The different load steps, each one including a static and a modal analysis, are performed within a *DO loop for simplicity.

```
! ** parameters
pi = acos(-1)
xa = 47.5e-3
xb = 16.5e-3
zh = 0.8e-3
spin = 120*2*pi

/prep7
et,1,181
r,1,zh

! ** material = aluminium
mp,ex,,7.2e+10
mp,nuxy,,.3
mp,dens,,2800.

! ** mesh
esize,0.0025
cyl4,,,xb,0,xa,360
amesh,all

! ** constraints = clamp inner radius
lsel,,,,5,8
dl,all,1,all
allsel
fini

! *** prestress modal analysis in rotation
nbstep = 5
dspin = spin/(nbstep-1)
*dim,spins,,nbstep
*vfill,spins,ramp,0.,dspin
spins(1) = 0.1     ! non zero to ease the Campbell diagram sorting
```

```
*do,iloop,1,nbstep

   /solu
   antype,static   ! Perform Static analysis.
   rescontrol, linear     ! Enable file writing for a subsequent linear
   coriolis,off,,,on  ! Coriolis effect is OFF in stationary reference
                ! frame
   omega,,,spins(iloop)
   campbell,rstp   ! Campbell for LP modal
   solve
   fini

   /solu
   antype,static, restart,,,perturb ! Perform a static restart with
                                    ! perturbation from the last load step and
                                    ! substep of the previous static solve
   perturb, modal,,, INERKEEP       ! Set the analysis options for perturbed
                                    ! modal analysis
   coriolis,on,,,on                 ! Coriolis effect is ON in stationary
                                    ! reference frame
   solve, elform                    ! Reform element matrices

   modopt,qrdamp,20,,,on
   mxpand,20
   omega,,,spins(iloop)
   solve
   fini
*enddo

! *** Campbell diagram
/post1
file,,rstp
/show,JPEG
plcamp
prcamp
finish
/show,CLOSE
```

7.3. Example: Modal Analysis Using ANSYS Workbench

ANSYS Workbench can be used to perform the modal analysis of a structure in rotation. The structure considered is the thin disk described in Example: Campbell Diagram Analysis (p. 45). The rotational velocity is 120 Hz.

In ANSYS Workbench, the settings are as follows:

- Analysis Settings > Solver Controls > Damped: Yes
- Analysis Settings > Rotordynamics Controls > Coriolis Effect: On

The rotational velocity is inserted as a load specifying the Magnitude and Axis of the Vector.

The following ANSYS input and output files were generated by the ANSYS Workbench product.

Input Listing
Output File

The mapped mesh of the disk is represented in Figure 7.4: Mapped Mesh of the Disk (p. 50)

Figure 7.4: Mapped Mesh of the Disk



The animation of the BW 2 nodal diameter mode is displayed in Figure 7.5: Animation of the Deformed Disk (p. 50)

Figure 7.5: Animation of the Deformed Disk



7.4. Example: Unbalance Harmonic Response of a Two-Spool Rotor

A sample input for the unbalance response of a two-spool rotor on symmetric bearings is located in Sample Unbalance Harmonic Analysis in the *Advanced Analysis Guide*.

7.5. Example: Mode-Superposition Harmonic Response to Base Excitation

The model, a cantilevered disk-spindle system, is shown in Figure 7.6: Cantilevered Disk Spindle (p. 51). The disk is fixed to the spindle with a rigid clamp and is rotating at 0.75*50 Hz. The base excitation is a harmonic force along the negative Y direction, with a frequency of up to 500 Hz.

Figure 7.6: Cantilevered Disk Spindle



7.5.1. Problem Specifications

The geometric properties of the disk are as follows:

Thickness: 1.0 mm
Inner radius: 0.1016 m
Outer radius: 0.2032 m

The geometric properties of the shaft are as follows:

Length: 0.4064 m
Radius: 0.0132 m

The clamp is modeled with constraint equations. The inertia properties of the clamp are:

Mass = 6.8748 kg

Inertia (XX,YY) = 0.0282 kg.m^2

Inertia (ZZ) = 0.0355 kg.m^2

The material properties for this analysis are as follows

Young's modulus (E) = 2.04e+11 N/m2
Poisson's ratio (υ) = 0.28

Density = 8030 kg/m^3

7.5.2. Input for the Analysis

```
! ** parameters
pi = acos(-1)
xb = 0.1016
xa = 0.2032
zh = 1.0e-3
rs = 0.0191
ls = 0.4064
d1 = 0.0132
spin = 50*2*pi*0.75
fexcit = 500

/prep7

! ** material
mp,ex,,2.04e+11
mp,nuxy,,.28
mp,dens,,8030.

! ** spindle
et,1,188
sectype,1,beam,csolid
secd,rs,30
type,1
secn,1
k,1,,,-ls-d1
k,2,,,-d1
l,1,2
lesize,1,,,5
lmesh,all

! ** disk
et,2,181
sectype,2,shell
secd,zh
type,2
secn,2
esize,0.01
cyl4,,,xb,0,xa,360
amesh,all

! ** clamp between disk and spindle
et,3,21
r,3,6.8748,6.8748,6.8748,0.0282,0.0282,0.0355
type,3
real,3
n,
ncent = node(0,0,0)
e,ncent
cerig,ncent,node(0,0,-d1),all
csys,1
nsel,,loc,x,xb
nsel,a,node,,ncent
cerig,ncent,all,all
allsel
csys,0

! ** constraints = clamp free end
nsel,,node,,node(0,0,-ls-d1)
d,all,all,0.0
allsel
fini

! *** modal analysis in rotation
/solu
antype,modal
modopt,qrdamp,30
mxpand,30
betad,1.e-5
coriolis,on,,,on
```

```
omega,,,spin
acel,,-1 !! generate load vector
solve
fini

! *** harmonic analysis in rotation
/solu
antype,harmonic
hropt,msup,30
outres,all,none
outres,nsol,all
acel,0,0,0
kbc,0
harfrq,,fexcit
nsubst,500
lvscale,1.0      !! use load vector
solve
fini

! *** expansion
/solu
expass,on
numexp,all
solve

! *** generate response plot
/post26
nsol,2,node(0,0,0),U,X,uxTip
nsol,3,node(0,0,0),U,Y,uyTip
nsol,4,node(0,xa,0),U,Z,uzDisk
/gropt,logy,on
/axlab,x,FREQUENCIES
/axlab,y,DISPLACEMENTS (m)
/show,JPEG
plvar,2,3,4
EXTREM,2,4,1
/show,CLOSE
```

7.5.3. Output for the Analysis

Figure 7.7: Output for the Cantilevered Disk Spindle (p. 53) shows the graph of displacement versus frequency.

Figure 7.7: Output for the Cantilevered Disk Spindle



7.6. Example: Mode-Superposition Transient Response to an Impulse

The model is depicted in Figure 7.8: Rotating Shaft (p. 54). The shaft is rotating at 105000 RPM and is supported by two bearings. It is excited by an impulse along the X axis at a node situated in the right overhung part of the rotor.

Figure 7.8: Rotating Shaft



7.6.1. Problem Specifications

The specifications for this model, including the geometry, and the stiffness characteristics for the identical bearings are found in VM247, "Campbell Diagrams and Critical Speeds Using Symmetric Bearings" in the *Mechanical APDL Verification Manual*.

7.6.2. Input for the Analysis

```
/PREP7

MP,EX,1,2.078e+11
MP,DENS,1,7806
MP,NUXY,1,0.3

et,1,273,,3      !! 3 circumferential nodes
nbdiam = 18
*dim,diam,array,nbdiam
diam(1) = 1.02e-2
diam(2) = 2.04e-2
diam(3) = 1.52e-2
diam(4) = 4.06e-2
diam(5) = diam(4)
diam(6) = 6.6e-2
diam(7) = diam(6)
diam(8) = 5.08e-2
diam(9) = diam(8)
diam(10) = 2.54e-2
diam(11) = diam(10)
diam(12) = 3.04e-2
diam(13) = diam(12)
diam(14) = 2.54e-2
diam(15) = diam(14)
diam(16) = 7.62e-2
diam(17) = 4.06e-2
```

```
diam(18) = diam(17)

k,1
k,2 ,diam(1)/2
k,3 ,diam(1)/2,1.27e-2
k,4 ,          ,1.27e-2
a,1,2,3,4

k,5 ,diam(2)/2,1.27e-2
k,6 ,diam(2)/2,5.08e-2
k,7 ,diam(3)/2,5.08e-2
k,8 ,          ,5.08e-2
a,4,3,5,6,7,8

k,9 ,diam(3)/2,7.62e-2
k,10,          ,7.62e-2
a,8,7,9,10

k,11,diam(4)/2,7.62e-2
k,12,diam(4)/2,8.89e-2
k,13,          ,8.89e-2
a,10,9,11,12,13

k,14,diam(5)/2,10.16e-2
k,15,          ,10.16e-2
a,13,12,14,15

k,16,diam(6)/2,10.16e-2
k,17,diam(6)/2,10.67e-2
k,18,3.04e-2/2,10.67e-2
k,19,          ,10.67e-2
a,15,14,16,17,18,19

k,20,diam(7)/2,11.43e-2
k,21,diam(8)/2,11.43e-2
k,22,3.56e-2/2,11.43e-2
k,23,3.04e-2/2,11.43e-2
a,18,17,20,21,22,23

k,24,diam(8)/2,12.7e-2
k,25,3.56e-2/2,12.7e-2
a,22,21,24,25

k,26,          ,12.7e-2
k,27,diam(9)/2,13.46e-2
k,28,diam(10)/2,13.46e-2
k,29,           ,13.46e-2
a,26,25,24,27,28,29

k,30,diam(10)/2,16.51e-2
k,31,           ,16.51e-2
a,29,28,30,31

k,32,diam(11)/2,19.05e-2
k,33,           ,19.05e-2
a,31,30,32,33

k,34,diam(12)/2,19.05e-2
k,35,diam(12)/2,22.86e-2
k,36,           ,22.86e-2
a,33,32,34,35,36

k,37,diam(13)/2,26.67e-2
k,38,diam(14)/2,26.67e-2
k,39,           ,26.67e-2
a,36,35,37,38,39

k,40,diam(14)/2,28.7e-2
k,41,           ,28.7e-2
a,39,38,40,41

k,42,diam(15)/2,30.48e-2
```

```
k,43,          ,30.48e-2
a,41,40,42,43

k,44,diam(16)/2,30.48e-2
k,45,diam(16)/2,31.5e-2
k,46,diam(17)/2,31.5e-2
k,47,          ,31.5e-2
a,43,42,44,45,46,47

k,48,diam(17)/2,34.54e-2
k,49,3.04e-2/2,34.54e-2
k,50,         ,34.54e-2
a,47,46,48,49,50

k,51,diam(18)/2,35.5e-2
k,52,3.04e-2/2,35.5e-2
a,49,48,51,52

esize,0.5e-2
amesh,all

sect,1,axis    !! symmetry axis along Y
secd,1, 0,0,0, 0,1,0
naxi

! bearings
et,3,combin14
keyopt,3,2,1
et,4,combin14
keyopt,4,2,2
et,5,combin14
keyopt,5,2,3
r,3,4.378e+7

visu = -0.02 !! visualization of bearing
n,10000,visu,16.51e-2
n,10001,visu,28.7e-2

type,3
real,3
e,node(0,16.51e-2,0),10000
e,node(0,28.7e-2,0),10001
type,4
real,3
e,node(0,16.51e-2,0),10000
e,node(0,28.7e-2,0),10001
type,5
real,3
e,node(0,16.51e-2,0),10000
e,node(0,28.7e-2,0),10001

d,10000,all
d,10001,all

fini

! *** modal analysis in rotation
pi = acos(-1)
spin = 105000*pi/30

/solu
antype,modal
modopt,qrdamp,10,1.0
coriolis,on,,,on
betad,1.e-5
omega,,spin
mxpand,10,,,yes
solve
fini

! *** mode-superposition transient analysis
dt = 1.0e-04
```

```
nodF = node(0.20300E-01,0.88900E-01,0)

/solu
antype,transient
trnopt,msup,10
deltim,dt
kbc,0
outres,all,none
outres,nsol,all
outres,rsol,all

f,nodF,FX,0
time,2*dt
solve

f,nodF,FX,1.e+3
time,10*dt
solve

f,nodF,FX,0
time,100*dt
solve
fini

! *** expansion pass
/solu
expass,on
numexp,all
solve
fini

! *** generate bearing reaction forces plot
/post26
rforce,2,10000,F,X,fxRightBearing
rforce,3,10000,F,Z,fzRightBearing
rforce,4,10001,F,X,fxLeftBearing
rforce,5,10001,F,Z,fzLeftBearing

/show,JPEG
plvar,2,3,4,5
EXTREME,2,5,1
/show,CLOSE
```

7.6.3. Output for the Analysis

The plot of Bearing Reaction Forces vs. Time is shown in Figure 7.9: Rotating Shaft Output (p. 58).

Figure 7.9: Rotating Shaft Output



7.7. Example: Transient Response of a Startup

The model is a simply supported shaft. A rigid disk is located at 1/3 of its length. A bearing is located at 2/3 of its length. The rotational velocity varies with a constant slope from zero at t = 0 to 5000 RPM at t = 4 s.

7.7.1. Problem Specifications

The geometric properties of the shaft are as follows:

Length: 0.4 m
Radius: 0.01 m

The inertia properties of the disk are:

Mass = 16.47 kg

Inertia (XX,YY) = 9.47e-2 kg.m^2

Inertia (ZZ) = 0.1861 kg.m^2

The material properties for this analysis are as follows:

Young's modulus (E) = 2.0e+11 N/m^2
Poisson's ratio (υ) = 0.3

Density = 7800 kg/m^3

The unbalance mass (0.1g) is located on the disk at a distance of 0.15 m from the center line of the shaft.

7.7.2. Input for the Analysis

```
/prep7
! ** parameters
length = 0.4
```

```
ro_shaft = 0.01
ro_disk = 0.15
md = 16.47
id = 9.427e-2
ip = 0.1861
kxx = 2.0e+5
kyy = 5.0e+5
beta = 2.e-4

! ** material = steel
mp,ex,1,2.0e+11
mp,nuxy,1,.3
mp,dens,1,7800

! ** elements types
et,1,188
sect,1,beam,csolid
secdata,ro_shaft,20
et,2,21
r,2,md,md,md,id,id,ip
et,3,14,,1
r,3,kxx,beta*kxx
et,4,14,,2
r,4,kyy,beta*kyy

! ** shaft
type,1
secn,1
mat,1
k,1
k,2,,,length
l,1,2
lesize,1,,,9
lmesh,all

! ** disk
type,2
real,2
e,5

! ** bearing
n,21,-0.05,,2*length/3
type,3
real,3
e,8,21
type,4
real,4
e,8,21

! ** constraints
dk,1,ux,,,,uy
dk,2,ux,,,,uy
d,all,uz
d,all,rotz
d,21,all
finish

! ** transient tabular force (unbalance)
pi = acos(-1)
spin = 5000*pi/30
tinc = 0.5e-3
tend = 4
spindot = spin/tend
nbp = nint(tend/tinc) + 1
unb = 1.e-4
f0 = unb*ro_disk

*dim,spinTab,table,nbp,,,TIME
*dim,rotTab, table,nbp,,,TIME
*dim,fxTab,  table,nbp,,,TIME
*dim,fyTab,  table,nbp,,,TIME
*vfill,spinTab(1,0),ramp,0,tinc
```

```
*vfill,rotTab(1,0), ramp,0,tinc
*vfill,fxTab(1,0),  ramp,0,tinc
*vfill,fyTab(1,0),  ramp,0,tinc
tt = 0
*do,iloop,1,nbp
   spinVal = spindot*tt
   spinTab(iloop,1) = spinVal
   spin2 = spinVal**2
   rotVal = spindot*tt**2/2
   rotTab(iloop,1) = rotVal
   sinr = sin(rotVal)
   cosr = cos(rotVal)
   fxTab(iloop,1)= f0*(-spin2*sinr + spindot*cosr)
   fyTab(iloop,1)= f0*( spin2*cosr + spindot*sinr)
   tt   = tt + tinc
*enddo
fini

! ** transient analysis
/solu
antype,transient

time,tend
deltim,tinc,tinc/10,tinc*10
kbc,0
coriolis,on,,,on
omega,,,spin
f,5,fx,%fxTab%
f,5,fy,%fyTab%
outres,all,all
solve
fini


! ** generate response graphs
/post26
nsol,2,5,U,X,UXdisk
prod,3,2,2
nsol,4,5,U,Y,UYdisk
prod,5,4,4
add,6,3,5
sqrt,7,6,,,Ampl_At_Disk
/axlab,y,Displacement (m)
/show,JPEG
plvar,7
EXTREME,7
/show,CLOSE

esol,8,4,5,smisc,32,Sy_At_Disk
esol,9,4,5,smisc,34,Sz_At_Disk
/axlab,y,Bending Stresses (N/m2)
/show,JPEG
plvar,8,9
EXTREME,8,9
/show,CLOSE
```

7.7.3. Output for the Analysis

Figure 7.10: Transient Response – Displacement vs. Time (p. 61) shows displacement vs. time.

Figure 7.10: Transient Response – Displacement vs. Time



Figure 7.11: Transient Response - Bending Stress vs. Time (p. 61) shows bending stress vs. time.

Figure 7.11: Transient Response - Bending Stress vs. Time



7.8. Example: Campbell Diagram Analysis of a Simple Rotor Supported by a CMS Superelement

A rotating shaft is supported by two symmetric bearings and a foundation structure. This foundation is reduced to a Component Mode Synthesis (CMS) superelement. A Campbell diagram analysis is performed on the model.

Figure 7.12: Rotor-Bearings-Foundation model



7.8.1. Problem Specifications

Problem Specifications:

Shaft length	= 0.595 m
Shaft diameter	= 0.015 m
Support block length	= 0.79 m

Support block height	= 0.25 m
Support block width	= 0.154 m
Support block thickness	= 0.025 m
Bearing stiffness	= 5E+07 N/m

The inertia properties of the disks are:

Disc No	Mass [kg]	Ixx [kg-m2]	Iyy [kg-m2]	Izz [kg-m2]
1	0.334	2.688e-4	2.688e-4	1.360e-4
2	0.782	7.02e-4	7.020e-4	4.310e-4
3	4.390	1.39e-2	1.194e-2	8.809e-3

The material properties for this analysis are:

Structure	Material	Young's modulus [N/m^2]	Density [kg/m^3]	Poisson's ratio
Shaft	Steel	2 E+11	7850	0.3
Support	Aluminum	7 E+10	2710	0.3

7.8.2. Input for the Analysis

```
!** parameters
fl=0.79    ! foundation length
fw=0.154   ! foundation width
ft=0.025   ! foundation thickness
ex_f=70e9   ! foundation young's modulus
dens_f=2710   ! foundation density
fsh=0.25   ! support height
fst=0.025   ! support thickness
sch=0.2    ! shaft centre height
bor=0.025   ! bearing or =0.025 m
bs=5e7    ! bearing stiffness n/m
!** disc 1
m1=0.334
j1=2.688e-4
i1=1.36e-4
!** disc 2
m2=0.782
j2=7.02e-4
i2=4.31e-4
!** disc 3
m3=4.39
ix3=1.39e-2
iy3=1.194e-2
iz3=8.809e-3
ex_s=200e9
dens_s=7850
s_or=0.015
```

```
/prep7

!** foundation
block,0,fw,0,ft,0,fl
block,0,fw,0,fsh,(0.23-fst/2),(0.23+fst/2)
block,0,fw,0,fsh,(0.668-fst/2),(0.668+fst/2)
wpcsys,-1,0
wpoff,fw/2,sch
cswpla,11,1,1,1,
cyl4, , ,bor, , , ,fl
vsbv,2,4,,,keep
vsbv,3,4,
vovlap,all
wpcsys,-1,0
csys,0
n,1,fw/2,0.2,0.073
n,2,fw/2,0.2,0.23
n,3,fw/2,0.2,0.23+0.147
n,4,fw/2,0.2,0.23+0.147+0.179
n,5,fw/2,0.2,0.23+0.147+0.179+0.112

!** shaft
et,1,188
keyopt,1,3,3
sectype,1,beam,csolid,shaft, 0
secoffset,cent
secdata,s_or
ex,1,ex_s
nuxy,1,0.3
dens,1,dens_s
r,1,
e,1,2
*repeat,4,1,1
et,2,21
r,2,m1,m1,m1,j1,i1,i1
r,3,m2,m2,m2,j3,i2,i2
r,4,m3,m3,m3,ix3,iy3,iz3
type,2    ! motor
real,4
e,1
real,2    ! disc 1
e,3
real,3    ! disc 2
e,4
et,5,185
keyopt,5,2,3
ex,5,ex_f
nuxy,5,0.3
dens,5,dens_f
r,5
type,5
mat,5
real,5
csys,11
lsel,s,loc,x,0.025
lesize,all,,,4
csys,0
lsel,s,loc,y,0.1*ft,0.9*ft
lesize,all,,,2
lsel,s,loc,z,(0.23-fst/3),(0.23+fst/3)
lesize,all,,,2,,1
lsel,s,loc,z,(0.668-fst/3),(0.668+fst/3)
lesize,all,,,2,,1
allsel
esize,0.05
vsweep,all
*get,maxnode,node,,num,max

!** bearings radial stiffness modeling
et,7,14
r,7,bs/4
type,7
```

```
real,7
csys,11
nsel,s,loc,x,0,bor
nsel,r,loc,z,0.23
e,2,node(bor,0,0.23)
e,2,node(bor,45,0.23)
e,2,node(bor,90,0.23)
e,2,node(bor,135,0.23)
e,2,node(bor,180,0.23)
e,2,node(bor,225,0.23)
e,2,node(bor,270,0.23)
e,2,node(bor,315,0.23)
nsel,s,loc,x,0,bor
nsel,r,loc,z,0.668
e,5,node(bor,0,0.668)
e,5,node(bor,45,0.668)
e,5,node(bor,90,0.668)
e,5,node(bor,135,0.668)
e,5,node(bor,180,0.668)
e,5,node(bor,225,0.668)
e,5,node(bor,270,0.668)
e,5,node(bor,315,0.668)
allsel

!** components
esel,s,ename,,188
esel,a,ename,,21
cm,rotor,elem    ! rotating parts
esel,a,ename,,14
nsle,s,1
cm,rotor_bear,elem   ! rotor+bearings (non se)
esel,inve
cm,support,elem    ! foundation (se)
allsel
save,rotor_supp_full,db
finish

!***************************
!  superelement generation
!***************************
/filname,support
/solu
antype,substr
ematwrite,yes
seopt,support,1,
cmsopt,fix,4
nsel,s,loc,x,bor
m,all,all
csys,0     ! fix bottom of the support structure
nsel,s,loc,y,0
d,all,all
cmsel,s,support,elem
nsle,s
solve     ! generate superelement "support.sub"
fini
save

!**********************************************
!  rotor+bearings+superelement modal analysis
!**********************************************
/clear, nostart
/filname, rotor_use
resume, rotor_supp_full,db  ! resume original assembly model
/prep7
vclear,all    ! delete foundation elements to be replaced by superelement
et,10,matrix50
type,10
se,support    ! read in superelement from file "support.sub"
allsel
/solu
pi=acos(-1)
rpmtorps=2*pi/60
```

```
antype,modal
modopt, qrdamp,6,1,0,1,on
mxpand,6, , ,1
coriolis,on,,,on
beta,1e-6
cmomega,rotor,0*rpmtorps,,,fw/2,sch,0.073,fw/2,sch,0.668
solve
cmomega,rotor,1000*rpmtorps,,,fw/2,sch,0.073,fw/2,sch,0.668
solve
cmomega,rotor,5000*rpmtorps,,,fw/2,sch,0.073,fw/2,sch,0.668
solve
cmomega,rotor,10000*rpmtorps,,,fw/2,sch,0.073,fw/2,sch,0.668
solve
save, rotor_use,db
finish

! **********************************
!  expand results for superelement
! **********************************
/clear, nostart
/filname, support
resume, support,db
/solu
expass,on
seexp, support, rotor_use,
numexp,all, , , yes
solve
finish

! **********************
!   review the results
! **********************
/clear, nostart
resume, rotor_use,db

/post1
cmsfile,clear
file, rotor_use,rst
/gropt,divx,5
/gropt,divy,5
/yrange,0,500,1
/show,jpeg
plcamp,on,1,rpm,,rotor  ! plot campbell diagram with 1st order excitation
prcamp,on,1,rpm,,rotor
/noerase
plcamp,on,2,rpm,,rotor  ! plot campbell diagram with 2nd order excitation
prcamp,on,2,rpm,,rotor
plcamp,on,3,rpm,,rotor  ! plot campbell diagram with 3rd order excitation
prcamp,on,3,rpm,,rotor
plcamp,on,4,rpm,,rotor  ! plot campbell diagram with 4th order excitation
prcamp,on,4,rpm,,rotor
/show,close
/erase
fini

!** combined results of foundation and rotor
/clear, nostart
/verify
resume, rotor_use,db
/post1
/eshape,1
cmsfile,clear
file, rotor_use,rst
set,3,2
/show,jpeg
plnsol,u,sum,2
*get, umax,plnsol,0,max
*stat,umax
/show,close
cmsfile,add,support.rst
set,last
esel,s,ename,,185
```

```
/show,jpeg
plnsol,u,sum,2
*get, umax,plnsol,0,max
*stat,umax
/show,close

finish
```

7.8.3. Outputs for the Analysis

Figure 7.13: Campbell Diagram



Figure 7.14: Mode Shape of the Rotor-Bearings-Foundation (results of the foundation superelement are not expanded)



Figure 7.15: Mode Shape of the Foundation Structure Alone



7.9. Example: Critical Speed Map Generation

The example consists of an eight stage centrifugal compressor supported by two tilting pad bearings. It is modeled with beam (BEAM188), bearing elements (COMBI214), and points masses (MASS21). APDLMath commands are used to directly calculate the critical speeds and generate the critical speed map.

7.9.1. Input for the Analysis

```
/filename,tutor-rag07s

/prep7
! *** material
mp, ex, 1, 1.33414e+11
mp, dens, 1, 7833.3834
mp, prxy, 1, 0.3

! *** rotor geometry
et, 1, beam188,,, 2
nbdiam = 34
*dim, diam, array, nbdiam
diam(1)  = 0.06985
diam(2)  = 0.110744
diam(3)  = 0.127
diam(4)  = diam(3)
diam(5)  = 0.1651
diam(6)  = 0.168148
diam(7)  = 0.157988
diam(8)  = 0.182372
diam(9)  = 0.17018
diam(10) = diam(9)
diam(11) = diam(9)
diam(12) = diam(9)
diam(13) = diam(9)
diam(14) = diam(9)
diam(15) = diam(9)
diam(16) = diam(9)
diam(17) = diam(9)
diam(18) = diam(9)
diam(19) = diam(9)
diam(20) = diam(9)
diam(21) = diam(9)
diam(22) = diam(9)
diam(23) = diam(9)
diam(24) = diam(9)
diam(25) = diam(9)
diam(26) = 0.1778
diam(27) = 0.168148
diam(28) = diam(27)
diam(29) = 0.1651
diam(30) = 0.127
diam(31) = diam(30)
diam(32) = 0.118872
diam(33) = 0.094996
diam(34) = 0.093472
*do,i,1,nbdiam
   sectype, i, beam, csolid
   secdata, diam(i)/2
*enddo

! *** point masses
et, 2, mass21
r,101,72.1224,72.1224,72.1224
r,102,3.22056,3.22056,3.22056
r,103,4.58136,4.58136,4.58136
r,104,2.54016,2.54016,2.54016
r,105,4.39992,4.39992,4.39992
r,106,5.57928,5.57928,5.57928
```

```
r,107,5.67,5.67,5.67
r,108,7.4844,7.4844,7.4844
r,109,6.66792,6.66792,6.66792
r,110,33.3396,33.3396,33.3396,0.1805617,0.09042717,0.09042717       ! disk
r,111,7.80192,7.80192,7.80192,0.00992065,0.00520907,0.00520907
r,112,33.3396,33.3396,33.3396,0.1805617,0.09042717,0.09042717       ! disk
r,113,7.80192,7.80192,7.80192,0.00992065,0.00520907,0.00520907
r,114,33.3396,33.3396,33.3396,0.1805617,0.09042717,0.09042717       ! disk
r,115,7.80192,7.80192,7.80192,0.00992065,0.00520907,0.00520907
r,116,36.69624,36.69624,36.69624,0.1805617,0.09042717,0.09042717    ! disk
r,117,13.19976,13.19976,13.19976
r,118,36.92304,36.92304,36.92304,0.1805617,0.09042717,0.09042717    ! disk
r,119,9.5256,9.5256,9.5256,0.01436885,0.00793067,0.00793067
r,120,34.33752,34.33752,34.33752,0.1805617,0.09042717,0.09042717    ! disk
r,121,9.5256,9.5256,9.5256,0.01436885,0.0079014,0.0079014
r,122,34.33752,34.33752,34.33752,0.1805617,0.09042717,0.09042717    ! disk
r,123,9.5256,9.5256,9.5256,0.01436885,0.00793067,0.00793067
r,124,36.56016,36.56016,36.56016,0.1805617,0.09042717,0.09042717    ! disk
r,125,17.14608,17.14608,17.14608,0.04799371,0.03540999,0.03540999
r,126,5.62464,5.62464,5.62464
r,127,5.94216,5.94216,5.94216
r,128,5.71536,5.71536,5.71536
r,129,5.53392,5.53392,5.53392
r,130,4.39992,4.39992,4.39992
r,131,1.99584,1.99584,1.99584
r,132,3.58344,3.58344,3.58344
r,133,7.39368,7.39368,7.39368
r,134,9.43488,9.43488,9.43488
r,135,4.71744,4.71744,4.71744

! *** bearing (YZ plane)
et, 3, 214
keyopt, 3, 2, 1
keyopt, 3, 3, 0

*dim, kb, array, 10, 1, 1
kb(1,1) = 1751181.1, 3772744.57, 8128282.20, 17511811.02, 37717971.02, 81282646.93, 175118110.24, 377280585.83, 8128

r,3, kb(1,1), kb(1,1)

! *** nodes (shaft along X)
n, 1 , 0.
n, 2 , 0.034544
n, 3 , 0.20574
n, 4 , 0.24384
n, 5 , 0.287274
n, 6 , 0.375412
n, 7 , 0.429768
n, 8 , 0.51943
n, 9 , 0.60579
n, 10, 0.668274
n, 11, 0.747014
n, 12, 0.809498
n, 13, 0.888238
n, 14, 0.950722
n, 15, 1.029462
n, 16, 1.091946
n, 17, 1.251966
n, 18, 1.411986
n, 19, 1.484122
n, 20, 1.575054
n, 21, 1.64719
n, 22, 1.738122
n, 23, 1.810258
n, 24, 1.90119
n, 25, 2.028444
n, 26, 2.10185
n, 27, 2.160524
n, 28, 2.250694
n, 29, 2.30505
n, 30, 2.39268
n, 31, 2.436114
```

```
n, 32, 2.474214
n, 33, 2.61874
n, 34, 2.666238
n, 35, 2.806192
n, 40 , 0.24384
n, 310, 2.436114

! *** mesh
type, 1
mat, 1
*do,i,1,nbdiam
   secnum, i
   e, i, i+1
*enddo
type, 2
*do,i,1,nbdiam+1
   real, i+100
   e, i
*enddo
type, 3
real, 3
e, 4, 40
e, 31, 310

! *** boundary conditions
d, all, ux ,,,,,rotx
d, 40,  all
d, 310, all
fini

! *** Example of how to determine critical speeds
!     -----------------------------------------
! ** modify the bearing stiffness
/prep7
rmodif, 3, 1, kb(5,1), kb(5,1)
finish

! ** write the full file
/solu
antype, modal
modopt, damp, 2
coriolis, on,,, on
omega, 1.0   ! unit rotational velocity to get [G1]
wrfull, 1         ! stop solve when the full file is written
solve
fini

! ** read matrices on the full file
*smat, K , D,IMPORT,FULL,tutor-rag07s.full, STIFF
*smat, M , D,IMPORT,FULL,tutor-rag07s.full, MASS
*smat, G1, D,IMPORT,FULL,tutor-rag07s.full, DAMP

! ** obtain the new eigenproblem matrices
alpha = 1.0
*smat, zMbar, Z, COPY, M
*axpy,, -1/alpha, G1, 1,, zMbar
*smat, zK, Z, COPY, K
*free,K
*free,M
*free,G1

! ** solve the new eigenproblem

/solu
antype, modal
modopt, unsym , 10
*eigen, zK, zMbar,, eigenVal, eigenVec
fini
*free,zK
*free,eigenVec

! ** store the critical speeds
```

```
*dim, critspeed1, ARRAY, 5
_coefunit = 60/alpha                        ! Hz -> rpm
*do,_iloop,1,5
   xx = eigenVal(_iloop*2,1)*_coefunit      ! real part of the eigenvalue (FW)
   critspeed1(_iloop) = xx
*enddo
*free,eigenVal
*status,critspeed1

! ** Obtain the complete critical speed map
!    Macro CRITSPEEDMAP.MAC
!       ARG1      Number of critical speeds
!       ARG2      Number of excitations per revolution
!                         =0  Defaults to 1.0 (synchronous excitation)
!       Bearings stiffness
!       ARG3      Lowest value
!       ARG4      Highest value
!       ARG5      Number of steps
!                         =0  Defaults to 10
!       Rotational velocity direction (normalized to unity)
!       ARG6      X component
!       ARG7      Y component
!       ARG8      Z component
!
!       Eigensolver parameter
!       ARG9      Beginning of the frequency range (defaults to 1 Hz)
!-------------------------------------------
/show,JPEG
critspeedmap, 5,, kb(1,1), kb(10,1),, 1,0,0
/show,CLOSE
```

7.9.2. Output for the Analysis

The critical speed map is shown in Figure 7.16: Critical Speed Map (p. 72).

Figure 7.16: Critical Speed Map



7.10. Example: Harmonic Response to Unbalanced Force using Component Mode Synthesis (CMS)

The following example shows a harmonic analysis with unbalanced force using the component mode synthesis (CMS) procedure on a rotating part. It illustrates the use of the **SYNCHRO** command and whirl animation using the **ANHARM** command during postprocessing.

The structure is a two-spool rotor on symmetric bearings as shown in Figure 7.17: Multi-Spool Rotor Model (Full and CMS Model) (p. 73). The outer spool rotates at up to 5000 RPM and the inner spool rotates about twice as fast.

This model is also used in Example Unbalance Harmonic Analysis in the *Advanced Analysis Guide*. In the following example, however, the inner spool is reduced to a CMS superelement.

Figure 7.17: Multi-Spool Rotor Model (Full and CMS Model)



7.10.1. Problem Specifications

An unbalanced force is applied at a node on the outer spool and the harmonic response is calculated.

7.10.2. Input for the Analysis

```
/out,
/com, ******************************************
/com, Multi-Spool CMS Model
/com, ******************************************
/out,scratch
/filname,cms_model
/prep7
et,1,281
type,1
sectype,1,shell
secdata,0.05/2,1,0,3

mp,ex,1,2.0e11
mp,dens,1,7800
```

```
mp,nuxy,1,0.3
mat,1

k,1,0,0.15,0
k,2,0,0.20,0
k,3,3,0.20,0
k,4,3,0.15,0

k,5,0,0,0
k,6,10,0,0

a,1,2,3,4

vrotat,1,,,,,,5,6,360

lsel,s,line,,2,8,2
lsel,a,line,,22,24,2
lsel,a,line,,14,16,2
lesize,all,,,6
lsel,all

lsel,s,line,,9,12,1
lsel,a,line,,17,20,1
lsel,a,line,,25,32,1
lesize,all,,,4
lsel,all

lsel,s,line,,1,7,2
lsel,a,line,,13,15,2
lsel,a,line,,21,23,2
lesize,all,,,1
lsel,all

amesh,all
allsel,all

et,3,281
type,3
secnum,3
mat,1
sectype,3,shell
secdata,0.05/2,1,0,7

k,35,7,0.20,0
k,36,7,0.15,0
k,37,10,0.15,0
k,38,10,0.20,0

a,35,36,37,38

vrotat,21,,,,,,5,6,360

lsel,s,line,,34,40,2
lsel,a,line,,54,56,2
lsel,a,line,,46,48,2
lesize,all,,,6
lsel,all

lsel,s,line,,57,64,1
lsel,a,line,,41,44,1
lsel,a,line,,49,52,1
lesize,all,,,4
lsel,all

lsel,s,line,,33,39,2
lsel,a,line,,45,47,2
lsel,a,line,,53,55,2
lesize,all,,,1
lsel,all

amesh,21,40,1
allsel,all
```

```
nummrg,node
nummrg,kp

numstr,node,3000
numstr,elem,3000

et,2,281
mp,ex,2,2.0e11
mp,dens,2,7800
mp,nuxy,2,0.3
mp,alpd,2,1e-3                  ! Material damping
mp,betd,2,1e-4
type,2
secnum,2
mat,2
sectype,2,shell
secdata,0.1/2,1,0,5

k,39,3,0.20,0
k,40,3,0.30,0
k,41,7,0.30,0
k,42,7,0.20,0

a,39,40,41,42

vrotat,41,,,,,,5,6,360

lsel,s,line,,66,72,2
lsel,a,line,,78,80,2
lsel,a,line,,86,88,2
lesize,all,,,8

lsel,s,line,,89,96,1
lsel,a,line,,81,84,1
lsel,a,line,,73,76,1
lesize,all,,,4

lsel,s,line,,65,71,2
lsel,a,line,,77,79,2
lsel,a,line,,85,87,2
lesize,all,,,1

amesh,41,60,1
allsel,all

cpintf,all

et,4,214
keyopt,4,2,1
keyopt,4,3,0
real,4
r,4,1.0e6,1.4e5,,,10,10
rmore,,,,

n,10000,5,0.22,0

type,4
real,4
e,node(5,0.20,0),10000
allsel,all

esel,s,type,,1
cm,comp1,elem
esel,all

esel,s,type,,2
cm,comp2,elem
esel,all

esel,s,type,,3
cm,comp3,elem
```

```
esel,all

nsel,s,loc,x,0
nsel,a,loc,x,10
d,all,all,0
nsel,all

d,10000,all,0

nsel,s,loc,x,0
cm,interface1,node
allsel,all,all

nsel,s,loc,x,3
cm,interface2,node
allsel,all,all

nsel,s,loc,x,7
cm,interface3,node
allsel,all,all

nsel,s,loc,x,10
cm,interface4,node
allsel,all,all

save
fini

/clear,nostart

/filname,model
/prep7
et,1,281
type,1
sectype,1,shell
secdata,0.05/2,1,0,3

mp,ex,1,2.0e11
mp,dens,1,7800
mp,nuxy,1,0.3

k,1,0,0.15,0
k,2,0,0.20,0
k,3,3,0.20,0
k,4,3,0.15,0

k,5,0,0,0
k,6,10,0,0

a,1,2,3,4

vrotat,1,,,,,,5,6,360

lsel,s,line,,2,8,2
lsel,a,line,,22,24,2
lsel,a,line,,14,16,2
lesize,all,,,6
lsel,all

lsel,s,line,,9,12,1
lsel,a,line,,17,20,1
lsel,a,line,,25,32,1
lesize,all,,,4
lsel,all

lsel,s,line,,1,7,2
lsel,a,line,,13,15,2
lsel,a,line,,21,23,2
lesize,all,,,1
lsel,all

amesh,all
```

```
allsel,all

et,3,281
type,3
secnum,3
mat,1
sectype,3,shell
secdata,0.05/2,1,0,7

k,35,7,0.20,0
k,36,7,0.15,0
k,37,10,0.15,0
k,38,10,0.20,0

a,35,36,37,38

vrotat,21,,,,,,5,6,360

lsel,s,line,,34,40,2
lsel,a,line,,54,56,2
lsel,a,line,,46,48,2
lesize,all,,,6
lsel,all

lsel,s,line,,57,64,1
lsel,a,line,,41,44,1
lsel,a,line,,49,52,1
lesize,all,,,4
lsel,all

lsel,s,line,,33,39,2
lsel,a,line,,45,47,2
lsel,a,line,,53,55,2
lesize,all,,,1
lsel,all

amesh,21,40,1
allsel,all

nummrg,kp
nummrg,node

esel,s,type,,1
cm,comp1,elem
esel,all

esel,s,type,,2
cm,comp2,elem
esel,all

esel,s,type,,3
cm,comp3,elem
esel,all

nsel,s,loc,x,0
cm,interface1,node
allsel,all,all

nsel,s,loc,x,3
cm,interface2,node
allsel,all,all

nsel,s,loc,x,7
cm,interface3,node
allsel,all,all

nsel,s,loc,x,10
cm,interface4,node
allsel,all,all

save
fini
```

```
!generation pass

/filname,part1
/sol
antype,substr
seopt,part1,3,1
cmsopt,fix,10

cmomega,comp1,1
coriolis,on,,,on                ! Coriolis on in a stationary reference frame

cmsel,s,comp1
cmsel,s,interface2
m,all,all
nsle

cmsel,s,interface1
m,all,all
nsle

solve
allsel,all,all
fini
save

/filname,part2
/sol
antype,substr
seopt,part2,3,1
cmsopt,fix,10

cmomega,comp3,1
coriolis,on,,,on                ! Coriolis on in a stationary reference frame

cmsel,s,comp3
cmsel,s,interface3
m,all,all
nsle

cmsel,s,interface4
m,all,all
nsle

solve
fini
save

!use pass

/clear,nostart
/filname,use
!/out,
/prep7
et,1,matrix50
type,1
se,part1
se,part2

numstr,node,3000
numstr,elem,3000

et,2,281
sectype,2,shell
secdata,0.1/2,1,0,5

mp,ex,1,2.0e11
mp,dens,1,7800
mp,nuxy,1,0.3
mp,alpd,1,1e-3                  ! Material damping
mp,betd,1,1e-4

type,2
```

```
mat,1
secnum,2

k,5,0,0,0
k,6,10,0,0

k,19,3,0.20,0
k,20,3,0.30,0
k,21,7,0.30,0
k,22,7,0.20,0

a,19,20,21,22

vrotat,1,,,,,,5,6,360

lsel,s,line,,2,8,2
lsel,a,line,,14,16,2
lsel,a,line,,22,24,2
lesize,all,,,8
lsel,all

lsel,s,line,,9,12,1
lsel,a,line,,17,20,1
lsel,a,line,,25,32,1
lesize,all,,,4
lsel,all

lsel,s,line,,1,7,2
lsel,a,line,,13,15,2
lsel,a,line,,21,23,2
lesize,all,,,1
lsel,all
nummrg,kp
amesh,1,20,1
allsel,all

nsel,s,loc,x,3
nsel,a,loc,x,7
cpintf,all
allse,all,all

et,4,214
keyopt,4,2,1
keyopt,4,3,0
real,4
r,4,1.0e6,1.4e5,,,10,10
rmore,,,,

n,10000,5,0.22,0

type,4
real,4
e,node(5,0.20,0),10000
allsel,all

nsel,s,loc,x,0
nsel,a,loc,x,10
d,all,all,0
nsel,all

d,10000,all,0

esel,s,type,,1
cm,comp1,elem
allsel,all,all

esel,s,type,,2
cm,comp2,elem
allsel,all,all

save
fini
```

```
/com, ******************************************
/com, Solution Controls for Full Harmonic Solve
/com, ******************************************
/solu
antype,harmic                    ! Perform Harmonic analysis
hropt,full

F0 = 1.0e-4                      ! Unbalance Force
n3 = node(5.0,-0.29424,0.58527e-1)

f,n3,fy,-F0                      ! Real FY component at node 'nodeUnb'
f,n3,fz,,F0                      ! Imaginary FZ component at node 'nodeUnb'

cmomega,comp1,1000
cmomega,comp2,500
coriolis,on,,,on                 ! Coriolis on in a stationary reference frame

synchro,,comp2

dmprat,0.02                      ! Global damping ratio

spinRpm1 = 0
spinRpm2 = 7200
begin_freq = spinRpm1/60         ! Begin frequency of excitation
end_freq = spinRpm2/60           ! End frequency of excitation
harfrq,begin_freq,end_freq
nsubs,120

kbc,1
solve
finish

/post1
file,use,rst

set,last
/show,jpeg
/graphics,power
/eshape,1
plnsol,u,sum
*get,umax,plnsol,0,max
/show,close
/out,
*stat,umax
/out,scratch
fini

!expansion pass

/clear,nostart
/filname,part1
resume,part1,db
/solu
expass,on
seexp,part1,use
numexp,all,,,yes
solve
fini

/clear,nostart
/filname,part2
resume,part2,db
/solu
expass,on
seexp,part2,use
numexp,all,,,yes
solve
fini

/inquire,test,exist,final,rst
*if,test,eq,1,then
/delete,final,rst
```

```
*endif

/clear,nostart
resume,cms_model,db

/post1
*do,j,1,120
file,part1                      ! generation pass 1 rst file
append,1,j
file,part2                      ! generation pass 1 rst file
append,1,j
file,use                        ! generation pass 1 rst file
append,1,j
reswrite,final,,,,1
file,part1                      ! generation pass 1 rst file
append,1,j,,1
file,part2                      ! generation pass 1 rst file
append,1,j,,1
file,use                        ! generation pass 1 rst file
append,1,j,,1

reswrite,final,,,,1
*enddo
fini

/post1
file,final,rst

set,last
/show,jpeg
/graphics,power
/eshape,1
plnsol,u,sum
*get,umax,plnsol,0,max
plesol,s,eqv
*get,smax,plnsol,0,max
plesol,epel,eqv
*get,emax,plnsol,0,max
/show,close
/out,
*stat,umax
*stat,smax
*stat,emax
/out,scratch

plnsol,u,sum
anharm,12,0.1,,,3,              ! Full Harmonic

fini

/post26
file,final,rst

n1 = node(5.0,0.58527e-1,0.29424)
n2 = node(5.0,0.29424,-0.58527e-1)

/com ***********************************************************************
/com, Output: Amplitude at nodes %n1% and %n2% as a function of the frequency
/com ***********************************************************************

nsol,2,n1,U,Y,UY
nsol,3,n1,U,Z,UZ
realvar,4,2,,,UYR
realvar,5,3,,,UZR
prod,6,4,4,,UYR_2
prod,7,5,5,,UZR_2
add,8,6,7,,UYR_2+UZR_2
sqrt,9,8,,,AMPL%n1%

extreme

nsol,2,n2,U,Y,UY
```

```
nsol,3,n2,U,Z,UZ
realvar,4,2,,,UYR
realvar,5,3,,,UZR
prod,6,4,4,,UYR_2
prod,7,5,5,,UZR_2
add,8,6,7,,UYR_2+UZR_2
sqrt,10,8,,,AMPL%n2%

extreme
/com *****************
/com, Graphics display
/com *****************
/gropt,logy,1
/show,jpeg
/title,'Unbalance Response Analysis using CMS Model'
!/yrange,1.0e-5,1.0e+2
/out,
plvar,9,10                      ! Displays variables in the form of a graph
/show,close
!prvar,9,10
finish

/exit,nosave
```

7.10.3. Output for the Analysis

The results of the unbalanced response analysis, postprocessed in **/POST26**, are shown in Figure 7.18: Unbalanced Response Using CMS Model (p. 83) and Figure 7.19: Unbalanced Response Using Full Model (p. 84). The logarithmic plots show the variation of the displacement amplitudes of two selected nodes with respect to the frequency of excitation.

Figure 7.18: Unbalanced Response Using CMS Model



Figure 7.19: Unbalanced Response Using Full Model



The animation of the whirls is shown below.



7.11. Example: Calculation of a Plain Cylindrical Journal Bearing Characteristics

The model is a plain cylindrical journal bearing supporting a shaft rotating at 100000 rad/s. The stiffness and damping characteristics are calculated at the adimensional equilibrium position XX = 0.5, YY = -0.7, and XXdot = YYdot = -0.05.

7.11.1. Problem Specifications

The bearing properties are as follows:

Clearance: 1 x 10^{-4} m
Length: 0.02 m
Radius: 0.01 m
Viscosity: 0.07 Pa·s

A perturbation increment of 1 x 10^{-5} is used.

7.11.2. Input for the Analysis

```
! ** Bearing parameters
mass    = 1      ! unused
xclear  = 1e-4
length  = 0.02
radius  = 0.01
mu      = 0.07
omegaj  = 1e+5
pertInc = 1e-5

! ** Equilibrium position (adimensional)
XX    = 0.5
YY    = -0.7
XXdot = -0.05
```

```
YYdot = -0.05

u1 = XX*xclear
u2 = YY*xclear
veloc1 = XXdot*xclear*omegaj
veloc2 = YYdot*xclear*omegaj

/prep7

! ** Nodes
n, 1, -1
n, 2,  0

! ** Elements

et, 1, 21,,, 2        ! 3D no rotary inertias
r, 1, mass

et, 2, 214
keyopt,2,1, 2         ! Reynolds integration
r, 2, xclear, length, radius, veloc1, veloc2, pertInc

! ** Material
mp,visc,2, mu

! ** Mesh
type,1
real,1
e, 2

type,2
real,2
mat,2
e, 1, 2

! ** Boundary conditions
d, all, all, 0.0
ddel, 2, UX
ddel, 2, UY
finish

! ** Static analysis with specified displacements
/solu
antype, static
omega,,, omegaj
d,2,ux, u1
d,2,uy, u2
outres,all,all
solve
finish

/post1
set,last

esel,,elem,,2

etable, fx, smisc,1
etable, fy, smisc,2

etable, theta1, nmisc,1
etable, theta2, nmisc,2
etable, mofp,   nmisc,3
etable, thetap, nmisc,4
etable, hmin,   nmisc,5
etable, thetah, nmisc,6

pretab

etable, kxx, nmisc,7
etable, kyy, nmisc,8
etable, kxy, nmisc,9
etable, kyx, nmisc,10
```

```
pretab, kxx, kyy, kxy, kyx

etable, cxx, nmisc,11
etable, cyy, nmisc,12
etable, cxy, nmisc,13
etable, cyx, nmisc,14

pretab, cxx, cyy, cxy, cyx
finish
```

7.11.3. Output for the Analysis

Figure 7.20: Bearing Element Results

```
***** POST1 ELEMENT TABLE LISTING *****

  STAT      CURRENT       CURRENT       CURRENT       CURRENT       CURRENT       CURRENT       CURR
  ELEM      FX            FY            THETA1        THETA2        MOFP          THETAP        HMIN
     2    17854.      -0.23311E+006 -233.13       -53.130       0.20445E+010  287.54        0.13977


***** POST1 ELEMENT TABLE LISTING *****

  STAT      CURRENT       CURRENT       CURRENT       CURRENT
  ELEM      KXX           KYY           KXY           KYX
     2   0.51068E+010  0.18480E+011-0.32363E+010-0.11633E+011


***** POST1 ELEMENT TABLE LISTING *****

  STAT      CURRENT       CURRENT       CURRENT       CURRENT
  ELEM      CXX           CYY           CXY           CYX
     2    29586.       0.13834E+006 -45399.       -45474.
```

7.12. Example: Calculation of a Squeeze Film Damper Characteristics

The model is a squeeze film damper supporting a shaft in circular precession motion. The stiffness and damping characteristics are calculated at the adimensional equilibrium position XX = 0.5. The precession rotational velocity is 500 rad/s.

7.12.1. Problem Specifications

The bearing properties are as follows:

Clearance: 2.54×10^{-5} m
Length: 0.005 m
Radius: 0.025 m
Viscosity: 0.07 Pa·s

A perturbation increment of 5×10^{-4} is used.

7.12.2. Input for the Analysis

```
/title, Squeeze Film Damper - Circular synchronous precession

! ** Bearing parameters
mass = 1      ! unused
xclear = 2.54e-5
length = 0.005
radius = 0.025
```

```
epsinc = 5e-4

mu = 0.07

! ** Equilibrium position
XX = 0.5
omegaprec = 500         ! precession

u1 = XX*xclear

/prep7

! ** Nodes
n, 1,  0
n, 2,  0
n, 3,  0

! ** Elements
et, 1, 21,,, 2        ! 3D no rotary inertias
r, 1, mass
type,1
real,1
e, 2

et, 2, 214
keyopt,2,1, 2         ! Reynolds integration
r, 2, xclear, length, radius, ,, epsinc
rmore,, omegaprec

! ** Material
mp,visc,2,mu

! ** Mesh
type,2
real,2
mat,2
e, 1, 2, 3

! ** Boundary conditions
d, all, all, 0.0
ddel, 2, UX
ddel, 2, UY
finish

! ** Static Analysis with specified displacement
/solu
antype, static
d,2,ux,u1
outres,all,all
solve
finish

/post1
set,last

esel,,elem,,2

etable, fx, smisc,1
etable, fy, smisc,2

etable, theta1, nmisc,1
etable, theta2, nmisc,2
etable, mofp,   nmisc,3
etable, thetap, nmisc,4
etable, hmin,   nmisc,5
etable, thetah, nmisc,6

pretab

etable, kxx, nmisc,7
etable, kyy, nmisc,8
etable, kxy, nmisc,9
```

```
etable, kyx, nmisc,10

pretab, kxx, kyy, kxy, kyx

etable, cxx, nmisc,11
etable, cyy, nmisc,12
etable, cxy, nmisc,13
etable, cyx, nmisc,14

pretab, cxx, cyy, cxy, cyx
finish
```

7.12.3. Output for the Analysis

Figure 7.21: Bearing Element Results

```
***** POST1 ELEMENT TABLE LISTING *****

  STAT     CURRENT      CURRENT      CURRENT      CURRENT      CURRENT      CURRENT
  ELEM     FX           FY           THETA1       THETA2       MOFP         THETAP
     2   149.28       203.74       0.0000       180.00      0.28040E+07  34.000       0

***** POST1 ELEMENT TABLE LISTING *****

  STAT     CURRENT      CURRENT      CURRENT      CURRENT
  ELEM     KXX          KYY          KXY          KYX
     2  0.39145E+08  0.0000       0.0000       0.0000

***** POST1 ELEMENT TABLE LISTING *****

  STAT     CURRENT      CURRENT      CURRENT      CURRENT
  ELEM     CXX          CYY          CXY          CYX
     2   0.0000       32085.       0.0000       0.0000
```

7.13. Example: Transient Analysis of a Plain Cylindrical Journal Bearing

This simple model is a plain cylindrical journal bearing supporting a point mass which represents a rotor. The 22.68 kg rotor is subjected to gravity and rotates at 4000 RPM. A transient analysis is performed for the first 5 rotational cycles (0.075 s). The maximum pressure in the bearing and the maximum amplitude of the bearing force are calculated and the rotor orbit is plotted.

7.13.1. Problem Specifications

The bearing properties are as follows:

Clearance: 1.27×10^{-4} (5 Mils)
Length: 0.0254 m (1 inch)
Radius: 0.0254 m (1 inch)
Viscosity: 6.89×10^{-2} Pa·s (1×10^{-5} Reyns)

7.13.2. Input for the Analysis

```
/title, Transient Analysis of a Plain Cylindrical Journal Bearing

! ** Rotor parameters
mass = 22.68
pi     = 4*atan(1)
omegaj = 4000*pi/30

! ** Bearing parameters
xclear = 1.27e-4
length = 0.0254
radius = 0.0254

mu = 6.89e-2

! ** Transient analysis parameters
nbcyc = 5
tend  = nbcyc*2*pi/omegaj
dt = 1e-4

/prep7

! ** Nodes
n, 1,  0
n, 2,  0
n, 3,  0

! ** Elements
et, 1, 21
r, 1, mass, mass, mass

et, 2, 214
keyopt,2,1, 2         ! Reynolds integration
r, 2, xclear, length, radius

! ** Material
mp,visc,2, mu

! ** Mesh
type,1
real,1
e, 2

type,2
real,2
mat,2
e, 1, 2, 3

! ** Boundary conditions
d, all, all, 0.0
ddel, 2, UX
ddel, 2, UY
fini

! ** Transient Analysis
/solu
antype, transient
nlgeom,on
acel,, 9.81
d, 2, OMGZ, omegaj
deltim, dt
time, tend
outres,all,all
solve
fini

/post26
numvar,20
nsol,2,2,u,x,ux2
nsol,3,2,u,y,uy2
```

```
esol,6,2,2,SMISC,1,fx
esol,7,2,2,SMISC,2,fy

esol,8,2,2,NMISC,3,mofp

prod,10,6,6,,fx_2
prod,11,7,7,,fy_2
add,12,10,11,,fx_2+fy_2
sqrt,13,12,,,fampl

prod,15,2,2,,ux2_2
prod,16,3,3,,uy2_2
add,17,15,16,,ux2_2+uy2_2
sqrt,18,17,,,ecc

extrem

*get,pmax,vari,8,extrem,vmax
*get,fmax,vari,13,extrem,vmax
*get,tfmax,vari,13,extrem,tmax
*get,eccend,vari,18,extrem,vlast

/show,png,rev
xvar,2
/axlab,x,ux@node2
/axlab,y,uy@node2
plvar,3
/reset
xvar,0
plvar,6,7
plvar,8

finish
```

7.13.3. Output for the Analysis

Figure 7.22: Result Parameters

```
*GET  PMAX     FROM  VARI    8  ITEM=EXTR VMAX  VALUE=  447744.375

*GET  FMAX     FROM  VARI   13  ITEM=EXTR VMAX  VALUE=  265.163705

*GET  TFMAX    FROM  VARI   13  ITEM=EXTR TMAX  VALUE= 0.460000000E-02

*GET  ECCEND   FROM  VARI   18  ITEM=EXTR VLAS  VALUE= 0.416683118E-04
```

Figure 7.23: Rotor Orbit Plot



Figure 7.24: Bearing Forces Plot



Figure 7.25: Maximum Fluid Film Pressure Plot



7.14. Example: Calculation of the Pressure Profile of a Plain Cylindrical Journal Bearing with Supply Orifice

The model is a plain cylindrical journal bearing with full leakage at both ends and rectangular supply orifice at 85° supporting a shaft rotating at 3000 RPM. The pressure profile is calculated at the adimensional equilibrium position XX = 0.4, YY = 0.4. The FLUID218 element is used in a static analysis. Element size is small to capture the variation of the pressure for each degree.

7.14.1. Problem Specifications

The bearing properties are as follows:

- Clearance: 2.5 x 10^{-5} m
- Length: 0.02 m
- Radius: 0.02 m
- Viscosity: 0.01 Pa˙s

Supply characteristics:

- Pressure: 0.5 x 10^6 Pa
- Orifice height: 10°
- Orifice width: ¼ of the diameter

7.14.2. Input for the Analysis

```
/title, Pressure Profile of a Plain Cylindrical Journal Bearing with Supply Orifice

! ** Main Parameters
rshaft       = 20e-3
lshaft       = 20e-3
omgshaft_rpm = 3000
XX           = 0.4
YY           = 0.4
xclear       = 25e-6
mu           = 0.01
Psup         = 0.5e+6

! ** Secondary Parameters
pi = 4*atan(1)
l2 = lshaft/2
l4 = lshaft/4
omgshaft = omgshaft_rpm*pi/30

/prep7
! ** SOLID185 (unused) - to support FLUID218 mesh only
et,1,185
mp,ex,1,1.0
mp,nuxy,1,1.0
mp,dens,1,1.0

! ** Solid Geometry and Mesh
type,1
mat,1
cylind,0,rshaft, -l2,l2, 0,90
ndvlz = 40      ! longitudinal lines divisions
lesiz,7,,,ndvlz
lesiz,8,,,ndvlz
lesiz,9,,,ndvlz
ndvrd = 12      ! radial lines divisions
lesiz,1,,,ndvrd
lesiz,6,,,ndvrd
lesiz,2,,,ndvrd
lesiz,5,,,ndvrd
ndvcr = 90      ! arc lines divisions
lesiz,3,,,ndvcr
lesiz,4,,,ndvcr
vsweep,all
vsymm,x,all
vsymm,y,all
nummrg,node
nummrg,kp

! ** FLUID218 Mesh
et,2,218
mp,visc,2, mu
r,2, xclear, rshaft, XX*xclear, YY*xclear
type,2
mat,2
real,2
csys,1
nsel,,loc,x,rshaft
esln
esurf
csys,0
allsel

! ** Remove Solid Elements
```

```
vclear,all
etdele,1

! ** Boundary Conditions
nsel, ,loc,z,-l2
nsel,a,loc,z,l2
d, all, pres, 0.0   ! zero pressure at both ends
allsel
csys,1
nsel, ,loc,x,rshaft
nsel,r,loc,y, 80,90
nsel,r,loc,z, -l4,l4
d, all, pres, PSup  ! supply pressure
csys,0
allsel
finish

! ** Static Analysis
/solu
antype,static
omega,,, omgshaft
outres,all,all
solve
fini

/post1
set,last
/show,png,rev
/view,,1,1,1

! ** Element Plot with Boundary Conditions
/pbc,pres,1
eplot
/pbc,pres,0

! ** Nodal Pressures
plnsol,pres

! ** Sum Element Forces
*get,nelem,ELEM,0,COUNT
fbx = 0
fby = 0
ielem = 0
*do,iloop,1,nelem
   ielem = ELNEXT(ielem)
   *get,con,ELEM,ielem,NMISC,11
   fbx = fbx + con
   *get,con,ELEM,ielem,NMISC,12
   fby = fby + con
*enddo
*status,fbx
*status,fby

! ** Element Fluid Velocities
rsys,solu
plesol,PG,X    ! tangential
plesol,PG,Y    ! axial
rsys
finish
```

7.14.3. Output for the Analysis

Figure 7.26: Bearing Forces

```
PARAMETER STATUS- FBX  (      23 PARAMETERS DEFINED)
                 (INCLUDING        2 INTERNAL PARAMETERS)

NAME                              VALUE                    TYPE  DIMENSIONS
FBX                               641.592908              SCALAR

PARAMETER STATUS- FBY  (      23 PARAMETERS DEFINED)
                 (INCLUDING        2 INTERNAL PARAMETERS)

NAME                              VALUE                    TYPE  DIMENSIONS
FBY                              -38.3821548              SCALAR
```

Figure 7.27: Element Plot with Pressure Boundary Conditions



Figure 7.28: Pressure Profile



Figure 7.29: Tangential Fluid Velocities at Mid-Thickness



Figure 7.30: Axial Fluid Velocities at Mid-Thickness



7.15. Example: Transient Analysis of a Plain Cylindrical Journal Bearing (3-D Approach)

The simple model of the bearing is described in Example: Transient Analysis of a Plain Cylindrical Journal Bearing (p. 89) where a 2-D approach is used. In this example, a 3-D model is created with SOLID185 and FLUID218 elements. The rotational velocity is applied to the shaft part using pilot nodes (MPC contacts) on the rear and front faces.

The mesh is coarse and a nonlinear large-deflection transient analysis is performed. In the first load step, the loads are applied gradually (**KBC**,0) and a very small time step is used to ease the convergence. Default force and pressure-based convergence criteria are specified (**CNVTOL**).

The variation of the position of the shaft center, bearing forces, maximum pressure, and minimum film thickness as a function of time is obtained.

7.15.1. Problem Specifications

See Problem Specifications (p. 89).

7.15.2. Input for the Analysis

```
/TITLE, Transient Analysis of a Plain Cylindrical Journal Bearing
! ** Main parameters
lshaft                = 2.54e-2
rshaft                = lshaft
mshaft              = 22.7
omgshaft_rpm = 4000
xclear              = 1.27e-4
mu                   = 0.069

! ** Secondary parameters
pi             = 4*atan(1)
l2             = lshaft/2
omgshaft = omgshaft_rpm*pi/30
roshaft     = mshaft/(lshaft*pi*rshaft**2)
zedge      = -l2

/prep7

! ** Solid Element
et,1,185
mp,ex,1,2.0e13
mp,nuxy,1,0.33
mp,dens,1,roshaft
mp,betd,1,0.1

! ** Geometry and Solid Mesh
cylind,0,rshaft, -l2,l2, 0  ,90
ndvlz = 10        ! longitudinal lines divisions
lesiz,7,,,ndvlz
lesiz,8,,,ndvlz
lesiz,9,,,ndvlz
ndvrd = 5         ! radial lines divisions
lesiz,1,,,ndvrd
lesiz,6,,,ndvrd
lesiz,2,,,ndvrd
lesiz,5,,,ndvrd
ndvcr = 10        ! arc lines divisions
lesiz,3,,,ndvcr
lesiz,4,,,ndvcr
type,1
mat,1
vsweep,all
vsymm,x,all
vsymm,y,all
nummrg,node
nummrg,kp

! ** MPC on Rear Face
*get,numn,NODE,0,NUM,MAX
numn = numn + 1
*get,nume,ELEM,0,NUM,MAX
nume = nume + 1
*set,tid,4
*set,cid,5
et,cid,174
et,tid,170
keyo,tid,2,1               ! Don't fix the pilot node
keyo,tid,4,111111
keyo,cid,12,5              ! Bonded Contact
keyo,cid,4,2               ! Rigid CERIG style load
keyo,cid,2,2               ! MPC style contact
type,cid
mat ,cid
real,cid
nsel,,loc,z,-l2
esln
esurf
*set,_npilot1,numn
n,_npilot1,0,0,-l2
```

```
type,tid
tshape,pilo
en,nume,_npilot1
tshape
allsel

! ** MPC on Front Face
*get,numn,NODE,0,NUM,MAX
numn = numn + 1
*get,nume,ELEM,0,NUM,MAX
nume = nume + 1
*set,tid,6
*set,cid,7
et,cid,174
et,tid,170
keyo,tid,2,1                ! Don't fix the pilot node
keyo,tid,4,111111
keyo,cid,12,5              ! Bonded Contact
keyo,cid,4,2                ! Rigid CERIG style load
keyo,cid,2,2                ! MPC style contact
type,cid
mat ,cid
real,cid
nsel,,loc,z,l2
esln
esurf
*set,_npilot2,numn
n,_npilot2,0,0,l2
type,tid
tshape,pilo
en,nume,_npilot2
tshape
allsel

! ** Bearing Element and Mesh
et,3,218
keyopt,3,1,1            !  U + PRES dofs
mp,visc,3,mu
r,3, xclear, rshaft
rmore, zedge
type,3
mat,3
real,3
csys,1
nsel,,loc,x,rshaft
esln
esurf
csys,0
allsel

! ** Boundary Conditions
nsel, ,loc,z,-l2
nsel,a,loc,z,l2
d, all, pres, 0.0d0      ! zero pressure at both ends
nsel, ,node,,_npilot1
nsel,a,node,,_npilot2
d, all, uz, 0.0d0,,,, rotx,roty    ! pilot nodes constraints
allsel
finish

! ** Transient Analysis
/solu
antype, transient
nlgeom,on
outres,all,all

nbdt1 = 10
dt1 = 1e-6
deltim, dt1
time, nbdt1*dt1
acel,, 9.81                        ! gravity
nsel, ,node,,_npilot1
```

```
nsel,a,node,,_npilot2
d, all, OMGZ, omgshaft   ! rotational velocity at pilot nodes
allsel
kbc,0
cnvtol,PRES              ! add default pressure criterion
cnvtol,F,,,,1.0            ! specify MINREF
cnvtol,FLOW,-1        ! remove fluid flow criterion
cnvtol,M,-1               ! remove moment criterion
solve

nbdt2 = 7500
dt2 = 1e-5
deltim, dt2
tend = nbdt1*dt1 + nbdt2*dt2
time, tend
kbc,1
solve
finish

/post1
/show,png,rev
/view,,1,1,1

! ** Bearing forces - Maximum Pressure - Minimum Thickness
nbdt = nbdt1 + nbdt2
*dim,fxtab,array,nbdt
*dim,fytab,array,nbdt
*dim,pmaxtab,array,nbdt
*dim,hmintab,array,nbdt
esel,,type,,3
*get,nelem,ELEM,0,COUNT
*do,iloops,1,nbdt

   *if,iloops,gt,nbdt1,then
      set,2,iloops-nbdt1
   *else
      set,1,iloops
   *endif

   f1 = 0
   f2 = 0
   pmax = 0
   hmin = xclear
   ielem = 0
   *do,iloop,1,nelem
      ielem = ELNEXT(ielem)
      *get,con,ELEM,ielem,NMISC,11
      f1 = f1 + con
      *get,con,ELEM,ielem,NMISC,12
      f2 = f2 + con
      *get,con,ELEM,ielem,NMISC,10
      *if,con,gt,pmax,then
         pmax = con
      *endif
      *get,con,ELEM,ielem,NMISC,9
      *if,con,lt,hmin,then
         hmin = con
      *endif
   *enddo
   fxtab(iloops)   = f1
   fytab(iloops)   = f2
   pmaxtab(iloops) = pmax
   hmintab(iloops) = hmin

*enddo
/out,
*status
/out,scratch
finish

/post26
nos = node(0,0,0)         ! shaft center
```

```
nsol,2,nos,u,x,uxs
nsol,3,nos,u,y,uys
plvar,2,3
xvar,2
/axlab,x,ux@node2
/axlab,y,uy@node2
plvar,3
/reset
xvar,0

vput,fxtab,4,0.0,,FX
vput,fytab,5,0.0,,FY
vput,pmaxtab,6,0.0,,PMAX
vput,hmintab,7,0.0,,HMIN

plvar,4,5
plvar,6
plvar,7
/out,
prvar,4,5,6,7
/out,scratch
finish

/post1
set,last
esel,,type,,3
plnsol,pres
finish
/exit,nosave
```

7.15.3. Output for the Analysis

Figure 7.31: Shaft Center Displacements



Figure 7.32: Shaft Center Orbit



Figure 7.33: Bearing Forces



Figure 7.34: Maximum Pressure (Elements Centroid Values)



Figure 7.35: Minimum Film Thickness (Elements Centroid Values)



Figure 7.36: Pressure Profile at End Time



7.16. Example: Quasi-Static Analysis of a Multi-Rotor System

The following example shows a quasi-static analysis of a two rotor system. In this analysis, the gyroscopic forces are included as a load vector. They are calculated from the gyroscopic matrix, generated from the rotational velocity input (**OMEGA** and **CMOMEGA**) and the activation of the **CORIOLIS** command, and from the instantaneous velocities, specified using **IC** and **ICROTATE** commands.

The structure is a two rotor system on symmetric bearings as shown in Figure 7.37: Multi-Rotor System (p. 111). The rotors, meshed using beam elements, are rotating at 1000 RPM and 2000 RPM respectively about its axis (global X-direction). The stator is meshed using shell elements. Parts are assembled using constraint equations (**CERIG** and **RBE3**). The whole structure is also rotating in a direction perpendicular to the shaft axis (global Z-direction), specified using the **ICROTATE** command.

Figure 7.37: Multi-Rotor System



The shafts nodes are rotated along the local z-direction aligned with the global X-direction. The instantaneous nodal velocities are specified along the local x-direction for both shafts using the **IC** command.

Figure 7.38: Rotation of Multi-Rotor System



7.16.1. Problem Specifications

Rotational velocity of shaft 1 = 1000 RPM
Rotational velocity of shaft 2 = 2000 RPM

The centrifugal force due to the **ICROTATE** command definition along the global Z-direction is included in the analysis.

7.16.2. Input for the Analysis

```
/out,scratch
/prep7

LOCAL,12,0,0,0,0,0,0,90
CSYS,0

! ** ROTOR 1

ET,1,188
SECTYPE,1,BEAM,CSOLID
SECDATA,0.1

MP,EX,1,2E11
MP,NUXY,1,0.3

k,1,0,0,0
k,2,9,0,0
l,1,2
lesize,all,,,9

TYPE,1
SECNUM,1
MAT,1
lmesh,all

CSYS,12
nrotate,all
CSYS,0

! ** ROTOR 2

ET,2,188
SECTYPE,2,BEAM,CSOLID
SECDATA,0.05

MP,EX,2,2E11
MP,NUXY,2,0.3

k,3,2,0,0
k,4,7
l,3,4
lesize,all,,,5

TYPE,2
SECNUM,2
MAT,2
lmesh,all

CSYS,12
nrotate,all
CSYS,0

! ** SHELL CASING

ET,3,181
SECTYPE,3,SHELL
SECDATA,0.1
```

```
MP,EX,3,1E11
MP,NUXY,3,0.3
MP,DENS,3,7800

CSYS,12
WPCSYS,-1
CYLIND,1,,1,3

TYPE,3
SECNUM,3
MAT,3
AMESH,3,4
CSYS,0
ALLSEL

! ** DISK MASS 1
ET,4,21
KEYOPT,4,2,0
R,4,75,75,75,125,125,250

TYPE,4
REAL,4
MAT,4

E,6
E,8
E,10

! ** DISK MASS 2
ET,5,21
KEYOPT,5,2,0
R,5,50,50,50,100,100,200

TYPE,5
REAL,5
MAT,5

E,13
E,15

! ** BEARING 1
N,1000,9,0,0
CSYS,12
nrotate,all
CSYS,0

ET,6,14
KEYOPT,6,2,1   ! >> along X direction
R,6,1E7

TYPE,6
REAL,6
MAT,6
E,2,1000

ET,7,14
KEYOPT,7,2,2  ! >> along Y direction
R,7,1E7

TYPE,7
REAL,7
MAT,7
E,2,1000

! ** BEARING 2
LSEL,S,,,3,6
NSLL
RBE3,3,ALL,ALL
ALLSEL

! ** BEARING 3
LSEL,S,,,7,10
```

```
NSLL
RBE3,13,ALL,ALL
ALLSEL

! ** BEARING 4
CERIG,9,12,UX,UY

! ** BOUNDARY CONDITIONS
D,1,ux
D,1,uy
D,1,uz
D,1,rotz

D,1000,ALL  ! >> grounded

! ** COMPONENTS DEFINITIONS
ESEL,S,TYPE,,1
ESEL,A,TYPE,,4
CM,ROT_1,ELEM
ALLSEL

ESEL,S,TYPE,,2
ESEL,A,TYPE,,5
CM,ROT_2,ELEM
ALLSEL,ALL,ALL
FINISH

! ** STATIC ANALYSIS
/solu
antype,static
coriolis,on,,,on

cmomega,ROT_1,1000,,,nx(1),ny(1),nz(1),nx(2),ny(2),nz(2)
cmomega,ROT_2,2000,,,nx(11),ny(11),nz(11),nx(12),ny(12),nz(12)

kbc,1

cmsel,s,ROT_1
nsle
ic,all,omgx,1.0   ! ** rotating about global Z (local x)
allsel,all

cmsel,s,ROT_2
nsle
ic,all,omgx,1.0   ! ** rotating about global Z (local x)
allsel,all

icrotate,all,1.0,4,0,0,4,0,1  ! ** axis of rotation passing though          ! ** node 6 along z - direction

outres,all,all
solve
finish

/post1
set,last
rsys,0
cmsel,s,ROT_1
cmsel,a,ROT_2
nsle
/out,
/com, ------------------------------------------------------------
/com, Response UX is due to the centrifugal force.
/com, Response UZ and ROTY are due to the gyroscopic moment.
/com, ------------------------------------------------------------
prnsol,dof
/out,scratch
/view,,1,1,1
/show,png,rev
plnsol,u,x  ! ** response due to centrifugal force
plnsol,u,z  ! ** response due to gyroscopic moment
/show,close
finish
```

```
/exit,nosave
```

7.16.3. Output for the Analysis

The results of the quasi-static analysis, postprocessed in **/POST1**, are shown in Figure 7.39: Displacement of the Shafts Along X-Direction Due to Centrifugal Force (p. 115) and Figure 7.40: Displacement of the Shafts Along Z-Direction Due to Gyroscopic Moment (p. 116).

Figure 7.39: Displacement of the Shafts Along X-Direction Due to Centrifugal Force



Figure 7.40: Displacement of the Shafts Along Z-Direction Due to Gyroscopic Moment



7.17. Example: Calculation of 3-D Hydrodynamic Bearing Characteristics

In this example, a 3-D model of a plain cylindrical journal bearing is created using FLUID218 element, supporting a shaft rotating at 10,000 RPM as shown in Figure 7.41: Model of a Plain Cylindrical Journal Bearing (p. 117). The stiffness and damping characteristics are calculated about the adimensional equilibrium position XX = 0.5, YY = -0.7, XXdot = YYdot = -0.05.

In order to evaluate the bearing characteristics, a small perturbation = 1×10^{-5} is given about the equilibrium position and the bearing forces are calculated. Using these bearing forces, the stiffness and damping characteristics are evaluated.

The bearing characteristics of the FLUID218 element model are compared with the bearing characteristics of the COMBI214 element model.

Figure 7.41: Model of a Plain Cylindrical Journal Bearing



7.17.1. Problem Specifications

The bearing properties are as follows:

- Clearance: 7.62 x 10^{-5} m
- Length: 1.143 x 10^{-2} m
- Radius: 6.477 x 10^{-2} m
- Viscosity: 2.632 x 10^{-3} Pa˙s

7.17.2. Input for the Analysis

```
/title, Calculation of a 3D Hydrodynamic Bearing Characteristics
/out,scratch
/com, ************************************************************************
/com,  Case 1: Force calculation @ equilibrium position (X ,Y, XXdot, YYdot)
/com, ************************************************************************
/filnam,case1
! Parameters (corresponding to non dimensional input)
pi      = 4*atan(1)
mshaft  = 1
xclear  = 7.62e-5
lshaft  = 0.01143
rshaft  = 0.06477
roshaft  = mshaft/(lshaft*pi*rshaft**2)    ! equivalent density
mu   = 0.002632
omegaj  = 1e4

pertInc  = 1e-5
XX     = 0.5
YY     = -0.7
XXdot  = -0.05
YYdot  = -0.05
veloc1  = XXdot*xclear*omegaj
veloc2  = YYdot*xclear*omegaj
zedge = -lshaft/2

/prep7

et,1,185   ! only used for FLUID218 meshing
mp,ex,1,1.0e11
```

```
mp,nuxy,1,mu
mp,dens,1,0.32051

cylind,0,rshaft,-lshaft/2,lshaft/2,0,90
cylind,0,rshaft,-lshaft/2,lshaft/2,90,180
cylind,0,rshaft,-lshaft/2,lshaft/2,180,270
cylind,0,rshaft,-lshaft/2,lshaft/2,270,360

nummrg,kp

esize,0.001

type,1
mat,1
vmesh,all

et, 2, 218
keyopt,2,1,0   ! PRES only dof
sectype,2,shell
secdata, xclear
r, 2, xclear, rshaft,XX*xclear,YY*xclear, veloc1, veloc2
rmore, zedge

mp,visc,2,mu
mp,dens,2,roshaft

type,2
mat,2
real,2

csys,1
nsel,,loc,x,rshaft
esln
esurf
csys,0
allsel

! --- remove solid elements
vclear,all
etdele,1

! --- boundary conditions
nsel,,loc,z,-lshaft/2
nsel,a,loc,z,lshaft/2
d, all, pres, 0.0d0      ! zero pressure at both ends
allsel

finish
save

/solu
antype, static
omega,,, omegaj
outres,all,all
solve
fini

/post1
file,case1,rst
set,last

esel,,type,,2
nsle

! sum the forces and moments to get total quantities
*get,nelem,ELEM,0,COUNT
f11 = 0
f12 = 0
ielem = 0
*do,iloop,1,nelem
   ielem = ELNEXT(ielem)
   *get,con,ELEM,ielem,NMISC,11
```

```
   f11 = f11 + con
   *get,con,ELEM,ielem,NMISC,12
   f12 = f12 + con
*enddo
allsel,all

/show,png,rev
/view,1,1,1,1
plnsol,pres
*get,presmax,plnsol,0,max
/show,close

fx   = f11
fy   = f12
mofp  = presmax
/out,
/com, ***********************************************************************
/com, Forces and Pressures @ equilibrium position using FLUID218 element model
/com, ***********************************************************************
*status,fx
*status,fy
*status,mofp
/out,scratch

parsav,all,temprm,parm
finish

/clear,nostart
/com, **************************************************************
/com,  Case 2: Force calculation @ equilibrium position with small
/com,   perturbation "dX" (X+dX, Y, XXdot, YYdot)
/com, **************************************************************
/out,scratch
/filnam,case2
/prep7
resume,case1,db
rdel,2

! Parameters (corresponding to non dimensional input)
dX   = pertInc
XXX    = XX + dX

r,2,xclear, rshaft,XXX*xclear,YY*xclear, veloc1, veloc2
rmore, zedge

fini
save

/solu
antype, static
omega,,, omegaj
outres,all,all
solve
fini

/post1
file,case2,rst
set,last

esel,,type,,2
nsle

! sum the forces and moments to get total quantities
*get,nelem,ELEM,0,COUNT
f21 = 0
f22 = 0
ielem = 0
*do,iloop,1,nelem
   ielem = ELNEXT(ielem)
   *get,con,ELEM,ielem,NMISC,11
   f21 = f21 + con
   *get,con,ELEM,ielem,NMISC,12
```

```
   f22 = f22 + con
*enddo
allsel,all

*status,f21
*status,f22
*status,presmax

parres,change,temprm,parm
parsav,all,temprm,parm
finish

/clear,nostart
/com, ***************************************************************
/com,  Case 3: Force calculation @ equilibrium position with small
/com,   perturbation "dY" (X,Y+dY, XXdot, YYdot)
/com, ***************************************************************
/out,scratch
/filnam,case3
/prep7
resume,case1,db
rdel,2

! Parameters (corresponding to non dimensional input)
dY    = pertInc
YYY   = YY + dY

r, 2, xclear, rshaft,XX*xclear,YYY*xclear, veloc1, veloc2
rmore, zedge

fini
save

/solu
antype, static
omega,,, omegaj
outres,all,all
solve
fini

/post1
file,case3,rst
set,last

esel,,type,,2
nsle

! sum the forces and moments to get total quantities
*get,nelem,ELEM,0,COUNT
f31 = 0
f32 = 0
ielem = 0
*do,iloop,1,nelem
   ielem = ELNEXT(ielem)
   *get,con,ELEM,ielem,NMISC,11
   f31 = f31 + con
   *get,con,ELEM,ielem,NMISC,12
   f32 = f32 + con
*enddo
allsel,all

*status,f31
*status,f32
*status,presmax

parres,change,temprm,parm
parsav,all,temprm,parm
finish

/clear,nostart
/com, **********************************************************************
/com,  Case 4: Force calculation @ equilibrium position with small
```

```
/com,   perturbation "dXXdot" (X, Y, XXdot+dXXdot, YYdot)
/com, *********************************************************************
/out,scratch
/filnam,case4
/prep7
resume,case1,db
rdel,2

! Parameters (corresponding to non dimensional input)
dXXdot  = pertInc
XXXdot   = XXdot + dXXdot
veloc1x  = XXXdot*xclear*omegaj

r, 2, xclear, rshaft,XX*xclear,YY*xclear, veloc1x, veloc2
rmore, zedge

fini
save

/solu
antype, static
omega,,, omegaj
outres,all,all
solve
fini

/post1
file,case4,rst
set,last

esel,,type,,2
nsle

! sum the forces and moments to get total quantities
*get,nelem,ELEM,0,COUNT
f41 = 0
f42 = 0
ielem = 0
*do,iloop,1,nelem
   ielem = ELNEXT(ielem)
   *get,con,ELEM,ielem,NMISC,11
   f41 = f41 + con
   *get,con,ELEM,ielem,NMISC,12
   f42 = f42 + con
*enddo
allsel,all

*status,f41
*status,f42
*status,presmax

parres,change,temprm,parm
parsav,all,temprm,parm
finish

/clear,nostart
/com, *********************************************************************
/com,  Case 5: Force calculation @ equilibrium position with small
/com,   perturbation of "dYYdot" (X,Y, XXdot, YYdot+dYYdot)
/com, *********************************************************************
/out,scratch
/filnam,case5
/prep7
resume,case1,db
rdel,2

! Parameters (corresponding to non dimensional input)
dYYdot  = pertInc
YYYdot   = YYdot + dYYdot
veloc2y  = YYYdot*xclear*omegaj

r, 2, xclear, rshaft,XX*xclear,YY*xclear, veloc1, veloc2y
```

```
rmore, zedge

fini
save

/solu
antype, static
omega,,, omegaj
outres,all,all
solve
fini

/post1
file,case5,rst
set,last

esel,,type,,2
nsle

! sum the forces and moments to get total quantities
*get,nelem,ELEM,0,COUNT
f51 = 0
f52 = 0
ielem = 0
*do,iloop,1,nelem
   ielem = ELNEXT(ielem)
   *get,con,ELEM,ielem,NMISC,11
   f51 = f51 + con
   *get,con,ELEM,ielem,NMISC,12
   f52 = f52 + con
*enddo
allsel,all

*status,f51
*status,f52
*status,presmax

parres,change,temprm,parm
parsav,all,temprm,parm
finish

/out,
/com, **************************************************
/com, Bearing characteristics of FLUID218 element model
/com, **************************************************
/out,scratch

parres,new,temprm,parm

/com, **************************************************
/com, Stiffness calculation
/com, **************************************************
dX = dX*xclear
dY = dY*xclear
KXX_ = (f21-f11)/dX
KYY_ = (f32-f12)/dY
KXY_ = (f31-f11)/dY
KYX_ = (f22-f12)/dX

/com, **************************************************
/com, Damping calculation
/com, **************************************************
dXXdot = dXXdot*xclear*omegaj
dYYdot = dYYdot*xclear*omegaj
CXX_ = (f41-f11)/dXXdot
CYY_ = (f52-f12)/dYYdot
CXY_ = (f51-f11)/dYYdot
CYX_ = (f42-f12)/dXXdot

/out,
/com, **************************************************
/com, Stiffness and Damping characteristics of FLUID218
```

```
/com, element model
/com, **************************************************
*status,PRM_

/clear,nostart
/out,
/com, *****************************************************
/com, Bearing characteristics of COMBI214
/com, *****************************************************
/out,scratch
/filnam,case6
! Resume parameters (corresponding to non dimensional input)
parres,new,temprm,parm

/prep7
n, 1, -1
n, 2,  0
n, 3,  0, -1

et, 1, 21,,,2         ! 3D mass, no rotary inertias
r, 1, mass,
type,1
real,1
e, 2

et, 2, 214
keyopt,2,1, 2          ! get outputs
r, 2, xclear, lshaft, rshaft, veloc1, veloc2,pertInc

mp,visc,2,mu

type,2
real,2
mat,2
e, 1, 2, 3

d, all, all, 0.0
ddel, 2, UX
ddel, 2, UY

fini
save

/solu
antype, static
omega,,, omegaj
d,2,ux,XX*xclear
d,2,uy,YY*xclear
outres,all,all
solve
fini

/post1
file,case6,rst
set,last

esel,,elem,,2

etable, fx, smisc,1
etable, fy, smisc,2
etable, mofp, nmisc,3

/out,
/com, ***********************************************************************
/com, Forces and Pressures @ equilibrium position using COMBI214 element model
/com, ***********************************************************************
pretab

/out,scratch

etable, kxx, nmisc,7
etable, kyy, nmisc,8
```

```
etable, kxy, nmisc,9
etable, kyx, nmisc,10

/out,
/com, ****************************************************
/com, Stiffness characteristics of COMBI214 element model
/com, ****************************************************
pretab, kxx, kyy, kxy, kyx
/out,scratch

etable, cxx, nmisc,11
etable, cyy, nmisc,12
etable, cxy, nmisc,13
etable, cyx, nmisc,14

/out,
/com, ****************************************************
/com, Damping characteristics of COMBI214 element model
/com, ****************************************************
pretab, cxx, cyy, cxy, cyx

finish
```

7.17.3. Output for the Analysis

Figure 7.42: Bearing Forces and Maximum Pressure Using FLUID218 Element Model

```
PARAMETER STATUS- FX    (      28 PARAMETERS DEFINED)
                       (INCLUDING        2 INTERNAL PARAMETERS)

NAME                                VALUE                        TYPE  DIMENSIONS
FX                                  1243.31122                  SCALAR

PARAMETER STATUS- FY    (      28 PARAMETERS DEFINED)
                       (INCLUDING        2 INTERNAL PARAMETERS)

NAME                                VALUE                        TYPE  DIMENSIONS
FY                                 -6195.26829                  SCALAR

PARAMETER STATUS- MOFP    (      28 PARAMETERS DEFINED)
                       (INCLUDING        2 INTERNAL PARAMETERS)

NAME                                VALUE                        TYPE  DIMENSIONS
MOFP                                24052920.7                  SCALAR
```

Figure 7.43: Bearing Characteristics Using FLUID218 Element Model

Stiffness Characteristics

```
PARAMETER STATUS- PRM_       (      54 PARAMETERS DEFINED)
                       (INCLUDING       10 INTERNAL PARAMETERS)

NAME                                VALUE                        TYPE  DIMENSIONS
KXX_                                276249021.                  SCALAR
KXY_                               -258970033.                  SCALAR
KYX_                               -613613916.                  SCALAR
KYY_                                971420277.                  SCALAR
```

Damping Characteristics

```
PARAMETER STATUS- PRM_      (     54 PARAMETERS DEFINED)
                  (INCLUDING      10 INTERNAL PARAMETERS)

NAME                            VALUE                   TYPE  DIMENSIONS
CXX_                            13832.5627             SCALAR
CXY_                           -23883.2974             SCALAR
CYX_                           -24265.2846             SCALAR
CYY_                            59453.7466             SCALAR
```

Figure 7.44: Bearing Forces and Maximum Pressure Using COMBI214 Element Model

```
PRINT ELEMENT TABLE ITEMS PER ELEMENT
  *****ANSYS VERIFICATION RUN ONLY*****
    DO NOT USE RESULTS FOR PRODUCTION

 ***** POST1 ELEMENT TABLE LISTING *****

   STAT     CURRENT       CURRENT       CURRENT
   ELEM     FX            FY            MOFP
      2   1313.2       -6439.5       0.24982E+008

MINIMUM VALUES
ELEM           0             0             0
VALUE    1313.2       -6439.5       0.24982E+008

MAXIMUM VALUES
ELEM           0             0             0
VALUE    1313.2       -6439.5       0.24982E+008
```

Figure 7.45: Bearing Characteristics Using COMBI214 Element Model

Stiffness Characteristics

```
PRINT ELEMENT TABLE ITEMS PER ELEMENT
  *****ANSYS VERIFICATION RUN ONLY*****
    DO NOT USE RESULTS FOR PRODUCTION

 ***** POST1 ELEMENT TABLE LISTING *****

   STAT     CURRENT       CURRENT       CURRENT       CURRENT
   ELEM     KXX           KYY           KXY           KYX
      2  0.29123E+009 0.10271E+010-0.27672E+009-0.64860E+009

MINIMUM VALUES
ELEM           0             0             0             0
VALUE   0.29123E+009 0.10271E+010-0.27672E+009-0.64860E+009

MAXIMUM VALUES
ELEM           0             0             0             0
VALUE   0.29123E+009 0.10271E+010-0.27672E+009-0.64860E+009
```

Damping Characteristics

```
PRINT ELEMENT TABLE ITEMS PER ELEMENT
  *****ANSYS VERIFICATION RUN ONLY*****
    DO NOT USE RESULTS FOR PRODUCTION

 ***** POST1 ELEMENT TABLE LISTING *****

   STAT     CURRENT      CURRENT      CURRENT      CURRENT
   ELEM     CXX          CYY          CXY          CYX
      2   13852.       60510.      -24214.      -24255.

MINIMUM VALUES
ELEM          0            0            0            0
VALUE    13852.       60510.      -24214.      -24255.

MAXIMUM VALUES
ELEM          0            0            0            0
VALUE    13852.       60510.      -24214.      -24255.
```

Appendix A. Bearing Characteristics File Format

Occurrence	Content	Format
1	Title 1	(A80)
1	Title 2	(A80)
1	Title 3	(A80)
1	nbstep, keyunit1 (optional), keyunit2 (optional)	(3I10)
nbstep	id, omega	(I10, F10.2)
	K11, K12, K21, K22 at omega	(4E15.7)
	C11, C12, C21, C22 at omega	(4E15.7)

Definitions:

nbstep -- Number of rotation velocity steps.

keyunit1 -- Key for the bearing characteristics units. If keyunit1 is not on file, it defaults to 1.

If keyunit1 = 0, there is no transformation.
If keyunit1 = 1, it transforms lbf/in into N/m and lbf.s/in into N.s/m.

keyunit2 -- Key for the rotational velocity unit. If keyunit2 is not on file, it defaults to 1.

If keyunit2 = 0, there is no transformation.
If keyunit2 = 1, it transforms rpm into rd/s.

id -- Step identification number.

omega -- Current step rotational velocity (see keyunit2 for units).

K11, K12,.. -- Stiffness characteristics at omega (see keyunit1 for units).

C11, C12,.. -- Damping characteristics at omega (see keyunit1 for units).

Index

S

T

U

W